
Introdução

Estas notas de aula são baseadas essencialmente em Wooldridge (2010), separando
o curso em duas partes, e uma avaliação para cada parte, denominadas de P1 e P2,
respectivamente.

A P1 versa sobre os capítulos 1 ao 10 do livro.

A P2 versa sobre os capítulos 13 ao 19 do livro.

Os códigos dos exercícios práticos estão em formato de Stata.

Sumário por característica da variável dependente

1. P1 (alfabetização)
1.1 LGN e TLC
1.2 Modelos lineares: y ∈ R

OLS, WLS, GLS
Endogeneidade e IV
Síntese para a P1 de 2026 Síntese para a P1 de 2026 (até o Cap. 7, não até o Cap. 10)
Painel sem dinâmica
Painel com dinâmica

2. P2 (métodos numéricos, máxima verossimilhança e modelos não lineares)
2.1 y ∈ {0, 1}
2.2 y ≥ mínimo
2.3 mínimo ≤ y ≤ máximo
2.4 y ∈ N
2.5 y ∈ {a, b, c}
2.6 y1 < y2 < y3
2.7 y em fronteiras

P1

4 / 221



LGN

cls
clear all
set seed 54321
set obs 1000
g d1 = ceil(runiform(0,6))
g d2 = ceil(runiform(0,6))
g s = d1+d2
g n = _n
g cum_amount = sum(s)
g cum_mean = cum_amount/n
scatter cum_mean n

cls
clear all
set seed 54321
set obs 1000
g x = rpoisson(2)
*g x = rbeta(1/2,1/2)
g n = _n
g cum_amount = sum(x)
g cum_mean = cum_amount/n
scatter cum_mean n

TLC

cls
clear all
set seed 54321
set obs 10000
sca a = 2
sca b = 4
sca m = 500
forv i = 1/`=m' {
g x`i' = a+(b-a)*runiform()

}
egen y = rowmean(x*)
g y_bar = (y-(a+b)/2)/(sqrt((b-a)^2/12)/sqrt(m))
kdensity y_bar

Modelos lineares

y ∈ R

▶ {(yit, yi0, xit, di)}n,T
i=1,t=1, em que y ∈ R, x ∈ R e d ∈ {0, 1}

y11 = γy10 + β1 + β2x11 + β3d11 + c1 + ε11
...

y1T = γy1(T −1) + β1 + β2x1T + β3d1T + c1 + ε1T

...
yn1 = γyn0 + β1 + β2xn1 + β3dn1 + cn + εn1

...
ynT = γyn(T −1) + β1 + β2xnT + β3dnT + cn + εnT

▶ ci é uma idiossincrasia constante em i entre 1 ≤ t ≤ T

▶ εit é um ruído aleatório

γ = 0 e T = 1

εit = ci + εit



y1 = β1 + β2x1 + β3d1 + u1

y2 = β1 + β2x2 + β3d2 + u2
...

yn = β1 + β2xn + β3dn + un

⇒


y1
y2
...

yn

 =


1 x1 d1
1 x2 d2
...

...
...

1 xn dn


β1

β2
β3

+


u1
u2
...

un



Y = Xβ + U ⇒ β̂ols = argminβ U ′U = (X ′X)−1(X ′Y )

9 / 221



MM

▶ Se ∄ correlação alguma entre quaisquer elementos de X e U , e a média do erro
é zero, ocorre E(X ′U) = 000

▶ O equivalente amostral de E(X ′U) é n−1X ′U =


∑n

i=1
ui
n∑n

i=1
xiui

n∑n
i=1

diui
n


▶ Y = Xβ + U ⇒ X ′Y = X ′Xβ + X ′U ⇒ X ′Y = X ′Xβ + n/nX ′U

▶ X ′Y
n→∞= X ′Xβ̂mm + 000⇒ (X ′X)−1X ′Y = (X ′X)−1X ′Xβ̂mm ⇒ β̂mm = β̂olsβ̂1

β̂2
β̂3

 =

 n
∑n

i=1 xi
∑n

i=1 di∑n
i=1 xi

∑n
i=1 x2

i

∑n
i=1 xidi∑n

i=1 di
∑n

i=1 dixi
∑n

i=1 d2
i


−1 

∑n
i=1 yi∑n

i=1 xiyi∑n
i=1 diyi



Base de dados hipotética

clear all
set seed 54321
set obs 100
sca b1 = -1
sca b2 = .5
sca b3 = .8
g x = rpoisson(9)
g c = rlogistic(.2,.9)
g z = rnormal(.9,.2)
g d = .4*c+.5*z+rnormal(0,1) > 0
g u = c+rnormal(0,1.2)
g y = round(b1+b2*x+b3*d+u,.01)
format y %9.2f
list y x d

Rotina padrão

g cte = 1
mkmat x d cte, matrix(X)
mkmat y, matrix(Y)

mat b_ols = inv(X'*X)*(X'*Y)
mat l b_ols

mat Y_hat = X*b_ols
mat u_hat = Y-Y_hat
mat SQR = u_hat'*u_hat
sca SQR = SQR[1,1]
sca s2 = SQR/(rowsof(X)-3)
mat V_beta = s2*inv(X'*X)
mat d_V_beta = vecdiag(V_beta)'

mat ep = J(3, 1, .)
mat t = J(3, 1, .)
forv i = 1/3 {
mat ep[`i',1] = d_V_beta[`i', 1]^.5
mat t[`i',1] = b_ols[`i',1]/ep[`i',1]

}
sum y
sca y_m = r(mean)
mat y_c = Y-y_m*J(100,1,1)
mat SQT = y_c'*y_c
sca SQT = SQT[1,1]
sca R2 = 1-SQR/SQT

reg y x d

Bootstrap como alternativa não paramétrica para “ep”

reg y x d, vce(bootstrap, dots(1))

cap program drop ols_boot
pro de ols_boot, rclass
preserve
bsample
mkmat cte x d, matrix(Xb)
mkmat y, matrix(Yb)
mat b_boot = inv(Xb'*Xb)*(Xb'*Yb)
ret sca b_1 = b_boot[1,1]
ret sca b_2 = b_boot[2,1]
ret sca b_3 = b_boot[3,1]
restore
end

cap ols_boot

di b_1

loc sim 50
mat results = J(`sim',3,.)
forv i = 1/`sim' {
cap ols_boot
mat results[`i', 1] = r(b_1)
mat results[`i', 2] = r(b_2)
mat results[`i', 3] = r(b_3)

}

mat list results
forv j = 1/3 {
g b_`j' = .
forv i = 1/`sim' {
replace b_`j' = results[`i', `j'] in `i'
}

}

sum b_*

Endogeneidade

▶ Um estimador β̂ é dito “consistente” para β se limn→∞ Pr(||β̂ − β|| < ϵ) = 1.
Nesse caso se escreve plim β̂ = β.

▶ Com um pouco de álgebra, temos que
β̂ols = β + (X ′X)−1(X ′U)⇒ β̂ols − β = (n−1(X ′X)−1)(n−1(X ′U)).

▶ Logo, plim β̂ols = β ⇐⇒ E(X ′U) = 000.
▶ Se E(X ′U) ̸= 000, dizemos que há endogeneidade no modelo.
▶ As variáveis que causam E(X ′U) ̸= 000 são ditas “endógenas”.

Razões da endogeneidade

▶ Variáveis Omitidas... em um modelo que busca explicar os salários com
base na educação, omitir a variável “habilidade inata” (que afeta tanto o
salário quanto o nível de educação) causaria endogeneidade.

▶ Causalidade Reversa... em um modelo onde se tenta prever o consumo com
base na renda, o consumo pode influenciar a renda, pois uma pessoa que
consome mais bens de luxo pode atrair oportunidades de negócios.

▶ Erros de Mensuração... em um modelo de leilão onde se deseja entender a
relação entre uma característica de um item (popularidade) e seu valor real.
Suponha que temos dados observacionais sobre o preço final de itens vendidos
em leilões, mas o preço não é uma medida perfeita do valor real. O preço,
neste contexto, pode ser influenciado por fatores temporários ou subjetivos,
que introduzem erros de mensuração na variável que representaria o valor.

Experimento para mostrar inconsistência

clear all

set seed 54321
set obs 1000

sca b1 = -1
sca b2 = .5
sca b3 = .8

g x = rpoisson(9)
g c = rlogistic(.2,.9)
g z = rnormal(.9,.2)
g d = .4*c+.5*z+rnormal(0,1)>0
g u = c+rnormal(0,1.2)
g y = b1+b2*x+b3*d+u

g b3_com_c = .
g b3_sem_c = .

forval j = 1/200 {
set seed `=54321`j''
replace x = rpoisson(9)
replace c = rlogistic(.2,.9)
replace z = rnormal(.9,.2)
replace d = .4*c+.5*z+rnormal(0,1) > 0
replace u = c+rnormal(0,1.2)
replace y = b1+b2*x+b3*d+u
reg y x d c
mat b_com_c = e(b)
replace b3_com_c = b_com_c[1,2] in `j'
reg y x d
mat b_sem_c = e(b)
replace b3_sem_c = b_sem_c[1,2] in `j'

}

twoway (kdensity b3_com_c) (kdensity b3_sem_c), ///
xline(.8, lpattern(dash) lcolor(blue)) xsize(16) ysize(9)

16 / 221



0
1

2
3

4

.5 1 1.5 2 2.5
x

kdensity b3_com_c kdensity b3_sem_c

17 / 221



A solução clássica é uma variável instrumental (IV) z que atende duas condições
(suponha d sendo a única variável endógena):

1. Relevância: Cov(z, d) ̸= 0.
2. Exogeneidade: Cov(z, u) = 0.

Z =


1 x1 z1
1 x2 z2
...

...
...

1 xn zn

⇒ β̂iv = (Z ′X)−1(Z ′U) = β + (Z ′X)−1(Z ′U)

▶ Mais de um instrumento ⇒ 2SLS.
▶ Desvantagens de IV:

▶ É impossível checar Cov(z, u) = 0, exigindo suposições teóricas.
▶ Um instrumento “fraco” pode causar mais viés que OLS.

Exemplos de instrumentos fracos

plim(β̂ols) = β1 + cov(X, u)
var(X) e plim(β̂iv) = β1 + cov(Z, u)

cov(Z, X)

▶ Exemplo 1
▶ β1 = 2, cov(X, u) = 1, var(X) = 4
▶ cov(Z, u) = 0.5, cov(Z, X) = 0.2
▶ Viés OLS: 0.25, Viés IV: 2.5

▶ Exemplo 2
▶ β1 = 1.5, cov(X, u) = 0.8, var(X) = 5
▶ cov(Z, u) = 2, cov(Z, X) = 1
▶ Viés OLS: 0.16, Viés IV: 2.0

▶ Conclusão: IV pode ser pior que OLS quando o instrumento é fraco.

19 / 221



mkmat cte x d, matrix(X)
mat l X
mkmat y, matrix(Y)
mat b_ols = inv(X'*X)*(X'*Y)
mat l b_ols
mkmat cte x z, matrix(Z)
mat b_iv = inv(Z'*X)*(Z'*Y)
mat l b_iv

reg y x d
est sto ols

ivregress 2sls y x (d = z)
est sto iv

estout ols iv, cells(b(star fmt(3)) se(par fmt(3)))

Principais testes com IV

Teste de Hausman
▶ Avaliar se as variáveis suspeitas de serem endógenas realmente o são.
▶ O teste compara as estimativas 2SLS com OLS.

Teste de Sargan
▶ Para selecionar instrumentos, quando há muitas possibilidades.

Teste de Stock-Yogo
▶ Para verificar a fraqueza dos instrumentos.
▶ Explora uma estatística F modificada do 1º estágio.

▶ Valores de teste surgem de um viés tolerável em relação ao viés do OLS.
▶ Por exemplo, se o F do 1º estágio é 3.65 e o valor crítico para uma tolerância de

25% é 5.53, considera-se o instrumento fraco.

Síntese para a P1 de 2026
(até o Cap. 7, não até o Cap. 10)

22 / 221



Parte I: modelo de salários e escolaridade

A equação estrutural de interesse é:

ln(wi) = β1 + β2 ln(xi) + β3si + β4Di + β5(Di ln xi) + ui.

▶ wi: salário.
▶ xi: experiência ou variável associada à experiência.
▶ si: escolaridade (anos de estudo).
▶ Di: indicador de grupo.
▶ Di ln(xi): permite retorno diferente de experiência entre grupos.

Parâmetro central: β3, o retorno da escolaridade. Usaremos si para evitar confusão
com a notação usual Z para matriz de instrumentos.

Por que si pode ser endógena?

Endogeneidade significa:

Cov(si, ui) ̸= 0 ou E(ui | si, controles) ̸= 0.

Possíveis fontes no problema:
▶ habilidade não observada;
▶ motivação;
▶ background familiar;
▶ qualidade da escola ou redes sociais não medidas.

Se essas variáveis afetam salários e também escolaridade, elas entram em ui e ficam
correlacionadas com si.

24 / 221



Consequência para MQO

No caso simples, o viés assintótico do MQO pode ser escrito como:

plim β̂MQO
3 = β3 + Cov(si, ui)

Var(si)
.

Com controles, a lógica é a mesma, mas usando a parte de si não explicada pelos
demais regressores.

Questão de consistência lógica
Se escolaridade está positivamente correlacionada com habilidade omitida e
habilidade aumenta salários, então Cov(si, ui) > 0 e o MQO tende a
superestimar β3.

Ideia de Proxy (da habilidade)

A proxy Pi pode reduzir viés de variável omitida se capturar parte da habilidade
não observada. Se incluirmos Pi, a ideia é reduzir a correlação entre escolaridade e
o erro:

Cov(si, ui | Pi, controles) ≈ 0.

Mas isso pode falhar porque a proxy pode ser fraca:

▶ a proxy mede habilidade com erro;
▶ habilidade é multidimensional;
▶ motivação e background familiar podem continuar omitidos;
▶ a proxy pode ser consequência de fatores já ligados à escolaridade e ao salário.

Ideia de Variáveis Instrumentais (da escolaridade)

Queremos uma fonte de variação em si que seja:
1. correlacionada com escolaridade;
2. não correlacionada com determinantes não observados dos salários.

Os instrumentos propostos são:
▶ Qi: exposição a reforma educacional que elevou a escolaridade obrigatória;
▶ Ri: proximidade geográfica, na adolescência, de uma instituição pública de

ensino médio.

Proxy versus IV/2SLS

Proxy Pi IV/2SLS com Qi, Ri

Habilidade Escolaridade
Controla diretamente parte do fa-
tor omitido.

Usa variação exógena em escolari-
dade.

Depende de boa mensuração da ha-
bilidade.

Depende de relevância e exclusão
dos instrumentos.

Pode reduzir viés, mas não neces-
sariamente eliminar.

Pode gerar consistência mesmo sem
observar habilidade.

Mantém interpretação mais pró-
xima do MQO condicionado.

Pode identificar efeito local para os
afetados pelos instrumentos.

Condições fundamentais de validade de instrumentos
Suponha dois candidatos a instrumentos... para Qi e Ri serem instrumentos
válidos:

1. Relevância

Cov(Qi, si | Xi) ̸= 0, Cov(Ri, si | Xi) ̸= 0.

Os instrumentos devem explicar escolaridade após controlar por ln(xi), Di e
Di ln(xi).

2. Exogeneidade / restrição de exclusão

Cov(Qi, ui) = 0, Cov(Ri, ui) = 0.

Eles só podem afetar salários por meio da escolaridade, não diretamente nem por
canais omitidos.

Notação matricial para IV/2SLS
Escreva o modelo estrutural como:

y = Xβ + u,

em que:
X =

[
W s

]
.

Aqui, W contém os regressores exógenos incluídos na equação estrutural:

W =
[
1 ln(x) D D ln(x)

]
,

e s é a variável potencialmente endógena, escolaridade. A matriz de instrumentos é:

Z =
[
W Q R

]
.

Ou seja, os instrumentos incluem os controles exógenos e os instrumentos excluídos
Q e R.

Demonstração: 2SLS como MQO sobre variáveis projetadas
No primeiro estágio matricial, projetamos os regressores em Z:

X̂ = PZX.

No segundo estágio, rodamos MQO de y em X̂:

β̂2SLS = (X̂ ′X̂)−1X̂ ′y.

Substituindo X̂ = PZX:

β̂2SLS = (X ′P ′
ZPZX)−1X ′P ′

Zy.

Como PZ é simétrica e idempotente, isto é, P ′
Z = PZ e PZPZ = PZ , então:

β̂2SLS = (X ′PZX)−1X ′PZy.

Conclusão formal: IV e 2SLS coincidem

Da demonstração anterior:

β̂2SLS = (X ′PZX)−1X ′PZy.

Essa é exatamente a forma matricial do estimador IV quando há mais
instrumentos do que variáveis endógenas.

Caso exatamente identificado
Se o número de instrumentos é igual ao número de regressores, a expressão se
reduz a:

β̂IV = (Z ′X)−1Z ′y.

Portanto, o 2SLS não é um estimador diferente do IV: ele é a forma operacional de
calcular IV por projeções lineares.

Como funciona o 2SLS neste problema

1. Estimar o primeiro estágio:

si = π1 + π2 ln(xi) + π3Di + π4(Di ln xi) + π5Qi + π6Ri + vi.

2. Obter o componente previsto ŝi, isto é, a parte da escolaridade explicada pelos
controles e pelos instrumentos.

3. Substituir si por ŝi na equação de salários:

ln(wi) = β1 + β2 ln(xi) + β3ŝi + β4Di + β5(Di ln xi) + εi.

Primeiro estágio

O primeiro estágio estimado foi:

ŝi = 8,20 + 0,35 ln(xi)− 0,25Di + 0,04(Di ln xi) + 0,90Qi + 0,15Ri.

Instrumento Coeficiente Erro-padrão Estatística t

Qi 0,90 0,28 3,21
Ri 0,15 0,06 2,50

Ambos têm associação positiva com escolaridade e são individualmente relevantes
em níveis convencionais.

Interpretação dos instrumentos no primeiro estágio

▶ π̂5 = 0,90: mantidos os controles constantes, exposição à reforma está
associada a 0,90 ano adicional de escolaridade, em média.

▶ π̂6 = 0,15: um aumento de uma unidade em proximidade geográfica está
associado a 0,15 ano adicional de escolaridade, em média.

As estatísticas t são:

tQ = 0,90
0,28 = 3,21, tR = 0,15

0,06 = 2,50.

Instrumentos fracos

Instrumentos fracos são instrumentos pouco correlacionados com a variável
endógena. Consequências:

▶ estimador IV pode ter grande viés em amostras finitas;
▶ erros-padrão podem ser pouco informativos;
▶ testes de hipótese podem ter tamanho distorcido;
▶ intervalos de confiança podem ser enganosos.

Por isso, a relevância não deve ser avaliada apenas instrumento por instrumento:
importa também a força conjunta dos instrumentos excluídos.

Teste F do primeiro estágio – Teste de Stock-Yogo
O teste conjunto dos instrumentos excluídos avalia:

H0 : π5 = π6 = 0.

No enunciado:
F = 9,40.

Regra prática usual
Um critério informal muito usado é: F > 10 sugere que os instrumentos não são
fracos. Aqui, F = 9,40 fica ligeiramente abaixo desse corte.

Stock-Yogo
Valores críticos de Stock-Yogo refinam essa análise, pois dependem do número de
instrumentos, número de variáveis endógenas e tolerância ao viés/tamanho do
teste.

Segundo estágio e semielasticidade

Segundo estágio:

l̂n(wi) = 0,95 + 0,44 ln(xi) + 0,11ŝi − 0,09Di + 0,07(Di ln xi).

O coeficiente de escolaridade é:

β̂2SLS
3 = 0,11.

Leitura econômica
Como a variável dependente está em log e si está em nível, 0,11 é uma
semielasticidade: um ano adicional de escolaridade aumenta o salário esperado em
aproximadamente 11%, mantendo os demais controles constantes.

MQO versus 2SLS

Estimador Coeficiente de escolaridade Erro-padrão
MQO 0,08 0,02
2SLS 0,11 0,05

O coeficiente IV é maior que o coeficiente MQO.

Interpretação possível
Isso sugere que o MQO pode estar subestimando o retorno da escolaridade, talvez
por erro de medida em escolaridade ou por heterogeneidade de retornos capturada
pelos instrumentos.
A conclusão exige cautela porque IV estima um efeito local associado à variação
induzida pelos instrumentos.

Teste de Hausman

O teste de Hausman compara MQO e IV:

H0 : MQO é consistente.

Resultado:
H = 5,80, p = 0,016.

Conclusão
Rejeita-se H0 em nível de 5%. Há evidência contra a exogeneidade de si e a favor
do uso de IV/2SLS.

Teste de Sargan

Como há dois instrumentos excluídos para uma variável endógena, o modelo é
sobreidentificado. O teste de Sargan avalia a compatibilidade conjunta dos
instrumentos com a restrição de exclusão:

H0 : instrumentos são válidos, isto é, exógenos.

Resultado:
J = 1,25, p = 0,264.

Conclusão
Não se rejeita H0. O teste não encontra evidência estatística contra a validade
conjunta dos instrumentos.
Atenção: não rejeitar não prova validade; apenas não detecta violação.

Parte II: heterocedasticidade e WLS

Modelo:
yi = β1 + β2xi + ui, i = 1, . . . , 4.

Amostra:
i 1 2 3 4
xi −1 1 −2 2
yi 1 3 −1 7

Heterocedasticidade conhecida:

Var(ui | xi) = σ2hi, h1 = h2 = 1, h3 = h4 = 4.

MQO: fórmulas

Com intercepto, os estimadores de MQO são:

β̂2 =
∑

i(xi − x̄)(yi − ȳ)∑
i(xi − x̄)2 , β̂1 = ȳ − β̂2x̄.

Neste exemplo:
x̄ = 0, ȳ = 1 + 3− 1 + 7

4 = 2,5.

Como x̄ = 0:
β̂2 =

∑
i xiyi∑
i x2

i

.

MQO: cálculo

∑
i

xiyi = (−1)(1) + (1)(3) + (−2)(−1) + (2)(7) = 18.

∑
i

x2
i = 1 + 1 + 4 + 4 = 10.

Logo:
β̂2 = 18

10 = 1,8, β̂1 = 2,5− 1,8(0) = 2,5.

Reta ajustada por MQO

ŷi = 2,5 + 1,8xi.

Por que usar WLS?

Sob heterocedasticidade conhecida:

Var(ui | xi) = σ2hi.

Observações com maior variância são menos precisas. O WLS minimiza:

∑
i

wi(yi − β1 − β2xi)2, wi = 1
hi

.

Neste exemplo:
w1 = w2 = 1, w3 = w4 = 1

4 .

WLS: médias ponderadas

A soma dos pesos é: ∑
i

wi = 1 + 1 + 1
4 + 1

4 = 2,5.

A média ponderada de x é:

x̄w =
∑

i wixi∑
i wi

=
−1 + 1− 1

2 + 1
2

2,5 = 0.

A média ponderada de y é:

ȳw =
∑

i wiyi∑
i wi

=
1 + 3− 1

4 + 7
4

2,5 = 5,5
2,5 = 2,2.

WLS: coeficientes
Como x̄w = 0:

β̂W LS
2 =

∑
i wixiyi∑
i wix2

i

.

Calculando: ∑
i

wixiyi = −1 + 3 + 1
2 + 14

4 = 6,

∑
i

wix
2
i = 1 + 1 + 1 + 1 = 4.

Logo:
β̂W LS

2 = 6
4 = 1,5, β̂W LS

1 = 2,2.

Reta ajustada por WLS

ŷW LS
i = 2,2 + 1,5xi.

Comparação numérica

Estimador Intercepto Inclinação
MQO 2,5 1,8
WLS 2,2 1,5

▶ MQO trata todas as observações como igualmente precisas.
▶ WLS reduz a influência das observações 3 e 4 porque elas têm variância

quatro vezes maior.
▶ Como as observações 3 e 4 têm |x| maior, a mudança de peso afeta

especialmente a inclinação.

Intuição dos pesos

Observações 3 e 4 têm:
Var(ui | xi) = 4σ2,

quatro vezes a variância das observações 1 e 2. Por isso, recebem peso:

w3 = w4 = 1
4 .

Ideia central
Se uma observação é mais ruidosa, ela contém menos informação sobre a reta
populacional. O WLS dá menos influência a essa observação no ajuste.

Eficiência: MQO ou WLS?

Sob as hipóteses:
E(ui | xi) = 0, Var(ui | xi) = σ2hi

e com hi corretamente especificado, MQO continua não viesado e consistente. Mas
WLS é mais eficiente, porque usa a informação sobre a estrutura da
heterocedasticidade.

MQO: resíduos
A reta ajustada por MQO foi:

ŷi = 2,5 + 1,8xi.

Logo:

i xi yi ŷi ûi = yi − ŷi

1 −1 1 0,7 0,3
2 1 3 4,3 −1,3
3 −2 −1 −1,1 0,1
4 2 7 6,1 0,9

A soma dos quadrados dos resíduos é:

SQRMQO = 0,32 + (−1,3)2 + 0,12 + 0,92 = 2,6.

MQO: cálculo do R2

A média de y é:
ȳ = 2,5.

A soma total dos quadrados é:

SQT =
∑

i

(yi − ȳ)2.

Logo:
SQT = (1− 2,5)2 + (3− 2,5)2 + (−1− 2,5)2 + (7− 2,5)2.

SQT = 2,25 + 0,25 + 12,25 + 20,25 = 35.

Portanto:
R2

MQO = 1− SQRMQO

SQT
= 1− 2,6

35 = 0,9257.

Interpretação
Neste exemplo, o MQO explica aproximadamente 92,6% da variação amostral de y.

MQO: variância residual
Como temos n = 4 observações e k = 2 parâmetros, os graus de liberdade são:

n− k = 4− 2 = 2.

A estimativa da variância do erro no MQO é:

σ̂2
MQO = SQRMQO

n− k
= 2,6

2 = 1,3.

Como:
X ′X =

[
4 0
0 10

]
,

temos:
(X ′X)−1 =

[
1/4 0
0 1/10

]
.

Logo:
V̂ar(β̂MQO) = σ̂2

MQO(X ′X)−1.

MQO: erros-padrão e estatísticas t
Assim:

V̂ar(β̂MQO) = 1,3
[
1/4 0
0 1/10

]
=
[
0,325 0

0 0,13

]
.

Os erros-padrão são:
se(β̂1) =

√
0,325 = 0,570,

se(β̂2) =
√

0,13 = 0,361.

As estatísticas t, para testar H0 : βj = 0, são:

tβ̂1
= 2,5

0,570 = 4,386,

tβ̂2
= 1,8

0,361 = 4,992.

Observação
Essas são as estatísticas t convencionais de MQO. Sob heterocedasticidade
conhecida, o WLS é o estimador mais adequado para inferência eficiente.

WLS: resíduos
A reta ajustada por WLS foi:

ŷW LS
i = 2,2 + 1,5xi.

Logo:

i xi yi wi ŷW LS
i ûW LS

i

1 −1 1 1 0,7 0,3
2 1 3 1 3,7 −0,7
3 −2 −1 1/4 −0,8 −0,2
4 2 7 1/4 5,2 1,8

No WLS, a soma relevante é a soma ponderada dos quadrados dos resíduos:

SQRPW LS =
∑

i

wi(ûW LS
i )2.

WLS: soma ponderada dos resíduos

Temos:
SQRPW LS = 1(0,3)2 + 1(−0,7)2 + 1

4(−0,2)2 + 1
4(1,8)2.

Logo:
SQRPW LS = 0,09 + 0,49 + 0,01 + 0,81 = 1,4.

A variância residual ponderada é:

σ̂2
W LS = SQRPW LS

n− k
= 1,4

2 = 0,7.

Comparação
A soma ponderada dos resíduos no WLS é menor porque o critério de minimização
agora leva em conta os pesos wi.

WLS: cálculo do R2 ponderado
A média ponderada de y foi:

ȳw = 2,2.

A soma total ponderada dos quadrados é:

SQTP =
∑

i

wi(yi − ȳw)2.

Logo:

SQTP = 1(1− 2,2)2 + 1(3− 2,2)2 + 1
4(−1− 2,2)2 + 1

4(7− 2,2)2.

SQTP = 1,44 + 0,64 + 2,56 + 5,76 = 10,4.

Portanto:
R2

W LS = 1− SQRPW LS

SQTP
= 1− 1,4

10,4 = 0,8654.

WLS: matriz de variância dos estimadores
No WLS:

V̂ar(β̂W LS) = σ̂2
W LS(X ′WX)−1.

Neste exemplo:

X ′WX =
[ ∑

i wi
∑

i wixi∑
i wixi

∑
i wix

2
i

]
.

Como: ∑
i

wi = 2,5,
∑

i

wixi = 0,
∑

i

wix
2
i = 4,

temos:
X ′WX =

[
2,5 0
0 4

]
.

Logo:

(X ′WX)−1 =
[
0,4 0
0 0,25

]
.

WLS: erros-padrão e estatísticas t
Como:

σ̂2
W LS = 0,7,

temos:
V̂ar(β̂W LS) = 0,7

[
0,4 0
0 0,25

]
=
[
0,28 0

0 0,175

]
.

Os erros-padrão são:
se(β̂W LS

1 ) =
√

0,28 = 0,529,

se(β̂W LS
2 ) =

√
0,175 = 0,418.

As estatísticas t, para testar H0 : βj = 0, são:

tβ̂W LS
1

= 2,2
0,529 = 4,158,

tβ̂W LS
2

= 1,5
0,418 = 3,586.

Comparação final: MQO e WLS

Estimador β̂1 se(β̂1) tβ̂1
β̂2 se(β̂2) tβ̂2

MQO 2,5 0,570 4,386 1,8 0,361 4,992
WLS 2,2 0,529 4,158 1,5 0,418 3,586

Estimador Soma de quadrados residual Variância residual R2

MQO 2,6 1,3 0,9257
WLS 1,4 ponderada 0,7 0,8654 ponderado

Mensagem
Os números de MQO e WLS não são diretamente comparáveis apenas pelo R2,
pois o WLS usa outro critério: ele minimiza a soma ponderada dos resíduos ao
quadrado.

Do WLS ao GLS
Até aqui, o problema era:

Var(ui | xi) = σ2hi.

Ou seja, cada observação podia ter uma variância diferente, mas os erros eram
tratados como não correlacionados entre si.
Em forma matricial, isso equivale a:

Var(u | X) = σ2


h1 0 0 0
0 h2 0 0
0 0 h3 0
0 0 0 h4

 .

Ideia
O WLS é suficiente quando a matriz de variância dos erros é diagonal. O GLS
aparece quando queremos permitir uma estrutura mais geral.

Motivação para GLS
O MQO é construído para o caso em que:

Var(u | X) = σ2I.

Isto significa duas coisas:

Var(ui | X) = σ2

e

Cov(ui, uj | X) = 0, i ̸= j.

Mas, em muitos problemas empíricos, essas duas hipóteses podem falhar.
▶ As variâncias podem ser diferentes entre observações.
▶ Os erros podem ser correlacionados entre observações.

Motivação
O GLS ajusta a estimação quando a matriz de variância-covariância dos erros não
é simplesmente σ2I. 62 / 221



Exemplos em que pode haver correlação dos erros
A correlação entre erros aparece naturalmente em vários contextos.
▶ Séries temporais: choques em um período podem persistir para períodos

seguintes.
▶ Dados em painel: observações da mesma firma, pessoa ou município podem

compartilhar componentes não observados.
▶ Dados espaciais: municípios vizinhos podem sofrer choques parecidos.
▶ Dados agrupados: estudantes da mesma escola ou trabalhadores da mesma

firma podem ter fatores comuns.

Por exemplo, em uma série temporal:

Cov(ut, ut−1) ̸= 0.

Problema
O MQO ignora essa dependência. O GLS usa essa informação para estimar de
forma mais eficiente.

Modelo geral do GLS
Escreva o modelo em forma matricial:

y = Xβ + u.

Agora suponha:

E(u | X) = 0,

mas:

Var(u | X) = σ2Ω.

A matriz Ω pode ter elementos diferentes na diagonal e fora da diagonal.

Ω =


ω11 ω12 · · · ω1n

ω21 ω22 · · · ω2n
...

... . . . ...
ωn1 ωn2 · · · ωnn

 .

Interpretação
A diagonal de Ω descreve as variâncias relativas. Os elementos fora da diagonal
descrevem as covariâncias entre os erros.

Estimador GLS
Se:

Var(u | X) = σ2Ω,

e Ω é conhecida, o estimador GLS é:

β̂GLS = (X ′Ω−1X)−1X ′Ω−1y.

Compare com o MQO:

β̂MQO = (X ′X)−1X ′y.

Diferença
O GLS substitui a matriz identidade por Ω−1. Assim, a estimação passa a levar em
conta a estrutura de variância e covariância dos erros.

GLS como minimização ponderada

O MQO minimiza:

(y −Xβ)′(y −Xβ).

O GLS minimiza:

(y −Xβ)′Ω−1(y −Xβ).

Ou seja, o GLS não mede o tamanho dos resíduos apenas pela soma dos
quadrados. Ele mede os resíduos levando em conta a matriz Ω−1.

Intuição
O GLS penaliza mais os resíduos que vêm de observações mais precisas e penaliza
menos os resíduos que vêm de observações mais ruidosas ou redundantes.

WLS como caso particular do GLS
No WLS, temos:

Ω =


h1 0 0 0
0 h2 0 0
0 0 h3 0
0 0 0 h4

 .

Logo:

Ω−1 =


1/h1 0 0 0

0 1/h2 0 0
0 0 1/h3 0
0 0 0 1/h4

 .

Portanto:

wi = 1
hi

.

Conclusão
O WLS é o GLS aplicado ao caso em que os erros têm variâncias diferentes, mas
não são correlacionados entre si.

Por que a correlação entre erros importa?

Suponha que duas observações tenham erros positivamente correlacionados:

Cov(ui, uj) > 0.

Isso significa que parte do erro de uma observação anda junto com o erro da outra.
Nesse caso, essas duas observações não trazem tanta informação independente
quanto pareceria no MQO.

Intuição
Se duas observações carregam choques parecidos, tratá-las como totalmente
independentes exagera a quantidade efetiva de informação na amostra.
O GLS corrige isso porque considera os termos de covariância dentro de Ω.

GLS como transformação do modelo
Se:

Var(u | X) = σ2Ω,

podemos encontrar uma matriz P tal que:

P ′P = Ω−1.

Multiplicando o modelo por P , temos:

Py = PXβ + Pu.

Defina:

y∗ = Py, X∗ = PX, u∗ = Pu.

Então:

y∗ = X∗β + u∗.

Resultado
No modelo transformado, o erro passa a ter variância proporcional à identidade. O
GLS é o MQO aplicado ao modelo transformado.

Variância do erro transformado
Como:

u∗ = Pu,

temos:

Var(u∗ | X) = Var(Pu | X).
Logo:

Var(u∗ | X) = P Var(u | X)P ′.

Como:

Var(u | X) = σ2Ω,

segue que:

Var(u∗ | X) = σ2PΩP ′.

Se a transformação é escolhida de modo que:

PΩP ′ = I,

então:

Var(u∗ | X) = σ2I.

Ideia
O GLS transforma o problema original em um problema no qual o MQO volta a
ser eficiente.

GLS factível

Na prática, raramente conhecemos exatamente Ω.
Por isso, muitas vezes usamos uma estimativa de Ω, denotada por Ω̂.
Nesse caso, o estimador é chamado de GLS factível:

β̂F GLS = (X ′Ω̂−1X)−1X ′Ω̂−1y.

Exemplo
Se suspeitamos de autocorrelação AR(1), podemos primeiro estimar o parâmetro
de autocorrelação e depois usar essa estimativa para construir Ω̂.
O mesmo raciocínio vale para heterocedasticidade modelada, dados agrupados ou
estruturas específicas de painel.

MQO robusto ou GLS?
Há duas estratégias diferentes diante de heterocedasticidade ou correlação dos
erros.

▶ MQO com erros-padrão robustos: mantém os coeficientes de MQO, mas
corrige a inferência.

▶ GLS ou FGLS: muda os coeficientes estimados, usando uma estrutura para
Ω.

Comparação
MQO robusto é mais defensivo: exige menos hipóteses sobre Ω. GLS é mais
eficiente se a estrutura de Ω estiver bem especificada.

Mensagem
O ganho do GLS depende de termos uma boa descrição da matriz de
variância-covariância dos erros.

Síntese: MQO, WLS e GLS

Estimador Estrutura dos erros Ideia central
MQO σ2I todos os erros têm mesma variância e não são correlacionados
WLS σ2 diag(hi) erros com variâncias diferentes, sem correlação
GLS σ2Ω variâncias diferentes e possíveis covariâncias entre erros

Conclusão
O WLS corrige um problema específico: heterocedasticidade conhecida. O GLS é a
versão geral: ele usa toda a matriz Ω para obter estimadores mais eficientes
quando os erros têm uma estrutura conhecida de variância e covariância.

Painel sem dinâmica

▶ γ = 0 e T > 1 

y11 = β1 + β2x11 + β3d11 + c1 + ε11

y12 = β1 + β2x12 + β3d12 + c1 + ε12
...

y1T = β1 + β2x1T + β3d1T + c1 + ε1T

...
yn1 = β1 + β2xn1 + β3dn1 + cn + εn1

yn2 = β1 + β2xn2 + β3dn2 + cn + εn2
...

ynT = β1 + β2xnT + β3dnT + cn + εnT

A abordagem da diferença da média (FE)


y11 − ȳ1 = β2(x11 − x̄1) + β3(d11 − d̄1) + (ε11 − ε̄1)
y12 − ȳ1 = β2(x12 − x̄1) + β3(d12 − d̄1) + (ε12 − ε̄1)

...
y1T − ȳ1 = β2(x1T − x̄1) + β3(d1T − d̄1) + (ε1T − ε̄1)

...
yn1 − ȳn = β2(xn1 − x̄n) + β3(dn1 − d̄n) + (εn1 − ε̄n)
yn2 − ȳn = β2(xn2 − x̄n) + β3(dn2 − d̄n) + (εn2 − ε̄n)

...
ynT − ȳn = β2(xnT − x̄n) + β3(dnT − d̄n) + (εnT − ε̄n)

β̂fe = (X̃ ′X̃)−1(X̃ ′Ỹ ), sem constante

A abordagem da primeira diferença (FD)



∆y11 = β2∆x11 + β3∆d11 + ∆ε11

∆y12 = β2∆x12 + β3∆d12 + ∆ε12
...

∆y1T = β2∆x1T + β3∆d1T + ∆ε1T

...
∆yn1 = β2∆xn1 + β3∆dn1 + ∆εn1

∆yn2 = β2∆xn2 + β3∆dn2 + ∆εn2
...

∆ynT = β2∆xnT + β3∆dnT + ∆εnT

⇒ β̂fd = (∆X ′∆X)−1(∆X ′∆Y )

clear all
set seed 54321
loc n = 20
loc T = 40
set obs `n'
g i = _n
expand `T'
bys i: gen t = _n
sca b1 = -1
sca b2 = .5
sca b3 = .8
sort i t
g x = rpoisson(9)
bys i: g c = rlogistic(.2,.9) if _n == 1
bys i: replace c = c[_n-1] if mi(c)
g z = rnormal(.9,.2)
g d = .4*c+.5*z+rnormal(0,1)>0
g u = c+rnormal(0,1.2)
g y = .
bys i: replace y = b1+b2*x+b3*d+u

xtset i t

foreach var in y x d {
bys i: egen `var'_mean = mean(`var')
g `var'_til = `var'-`var'_mean
g `var'_dif = D.`var'

}

mkmat x_til d_til, matrix(X_til)
mkmat y_til, matrix(Y_til)
mat beta_fe = inv(X_til'*X_til)*(X_til'*Y_til)
mat l beta_fe

drop if y_dif == .

mkmat x_dif d_dif, matrix(X_dif)
mkmat y_dif, matrix(Y_dif)
mat beta_fd = inv(X_dif'*X_dif)*(X_dif'*Y_dif)
mat l beta_fd

▶ Desvantagens da abordagem, independentemente de FE ou FD:
▶ Só resolve a endogeneidade se o problema for relacionado com uma variável

omitida que não muda com o tempo.
▶ Essa ideia é estranha se T é grande.
▶ Mesmo para T pequeno, a abordagem só resolve a endogeneidade se o problema

que a gera é constante para todas as unidades de observação.
▶ Não resolve problemas de causalidade reversa ou erro de medida.
▶ O efeito de interesse não pode estar relacionado com uma constante, já que há

perda de informação com variáveis constantes.
▶ FE/FD podem ser usados facilmente com IV.
▶ FD tem 2 desvantagens em relação a FE:

▶ Perde n observações
▶ Problemas em painel desbalanceado.

A abordagem de efeitos aleatórios (RE)
Se X não possui elementos endógenos, OLS é consistente e FE/FD são desnecessários, mas
a matriz de variância-covariância do erro precisa ser ajustada:

▶ uit = ci + εit ⇒ σ2
u = σ2

c + σ2
ε ⇒ UU ′

n=3,T =2 =


2.9 0.4 0 0 0 0
0.4 2.9 0 0 0 0
0 0 2.9 0.4 0 0
0 0 0.4 2.9 0 0
0 0 0 0 2.9 0.4
0 0 0 0 0.4 2.9


▶ É o caso de se usar β̂gls = (X ′Ω−1X)−1(X ′Ω−1Y ).

▶ β̂re = (X ′Ω−1X)−1(X ′Ω−1Y ) = (X̃ ′X̃)−1X̃ ′ỹ.
▶ Em que ỹit = yit − θȳi para θ = 1− σε/

√
Tσ2

c + σ2
ε .

▶ Se X não possui elementos endógenos, este GLS seria eficiente em detrimento ao
OLS e ao FE, embora todos sejam consistentes.

▶ O procedimento se dá em 2 estágios.
▶ RE subsidia o teste de Hausman.

sort i t
xtset i t
reg y x d
est sto ols
xtreg y x d, pa corr(independent)
est sto pols
xtreg y x d, fe
est sto fe
reg D.y D.x D.d, nocon
est sto fd
xtivreg y x (d = z), fe
est sto ivfe
xtreg y x d, re
est sto re
estout ols pols fe fd re ivfe, cells(b(star fmt(3)) se(par fmt(3)))
hausman fe re, sigmamore

------------------------------------------------------------------------------------------------------------
ols pols fe fd re ivfe

b/se b/se b/se b/se b/se b/se
------------------------------------------------------------------------------------------------------------
x 0.559*** 0.559*** 0.523*** 0.524*** 0.520***

(0.030) (0.030) (0.015) (0.016) (0.018)
d 3.548*** 3.548*** 0.808*** 0.909*** 0.101

(0.191) (0.191) (0.117) (0.120) (1.731)
D.x 0.514***

(0.015)
D.d 0.735***

(0.112)
_cons -2.880*** -2.880*** -0.658*** -0.738* -0.136

(0.314) (0.313) (0.166) (0.333) (1.286)
------------------------------------------------------------------------------------------------------------

---- Coefficients ----
| (b) (B) (b-B) sqrt(diag(V_b-V_B))
| fe re Difference Std. err.

-------------+----------------------------------------------------------------
x | .5233016 .5244211 -.0011196.0004307
d | .8082739 .9092188 -.1009449.0134717

------------------------------------------------------------------------------
b = Consistent under H0 and Ha; obtained from xtreg.
B = Inconsistent under Ha, efficient under H0; obtained from xtreg.

Test of H0: Difference in coefficients not systematic

chi2(2) = (b-B)'[(V_b-V_B)^(-1)](b-B)
= 57.43

Prob > chi2 = 0.0000

O p-valor 0 indica que se deve rejeitar a hipótese nula, o que significa que FE é preferível.

Acrescentando a dinâmica

Raíz unitária... só faz sentido painel dinâmico se |γ| < 1

yit = γyit−1 + ci + εit ⇒ yit = γtyi0 + 1− γt

1− γ
ci +

t−1∑
j=0

γjεit−j

A lógica do ADF segue sendo usado (e aqui a questão do balanceamento do painel
importa, por conta da FD): ∆yit = ρyit−1 +∑p

j=1 αj∆yit−j + ruídoit

Testes populares H0: ρ = 0 H1
LLC Raiz unitária ∀i Estacionariedade ∀i
IPS Raiz unitária ∀i Estacionariedade ̸ ∀i

Fisher Raiz unitária ∀i Estacionariedade ̸ ∀i

O p-valor pequeno rejeita H0 de que as séries contêm uma raiz unitária.

clear all
set seed 54321
loc n = 1000
loc T = 20
set obs `n'
g i = _n
expand `T'
bys i: gen t = _n
sca gamma = .6
sca b1 = -1
sca b2 = .5
sca b3 = .8
g x = rpoisson(9)

sort i t
bys i: g c = rlogistic(.2,.9) if _n == 1
bys i: replace c = c[_n-1] if mi(c)
g z = rnormal(.9,.2)
g d = .4*c+.5*z+rnormal(0,1)>0
g u = c+rnormal(0,1.2)
g y = .
bys i: replace y = b1+b2*x+b3*d+u if t == 1
bys i: replace y = gamma*y[_n-1]+b1+b2*x+b3*d+u if t > 1

xtset i t

* Teste Levin-Lin-Chu (LLC) com 1 defasagem
xtunitroot llc y, lags(1)

* Teste Im-Pesaran-Shin (IPS) com 2 defasagens
xtunitroot ips y, lags(2)

* Teste Fisher com 1 defasagem
xtunitroot fisher y, dfuller lags(1)

OLS, FE, FD etc. são inconsistentes em painel dinâmico

O procedimento de FE:

yit − ȳi = γ(yit−1 − ȳi,−1) + (εit − ε̄i)

Em que ȳi,−1 = (yi0 + yi1 + ... + yit−1)/T .

γ̂fe =
∑n

i=1
∑T

t=1(yit−1 − ȳi,−1)(yit − ȳi)∑n
i=1

∑T
t=1(yit−1 − ȳi,−1)2

=
∑n

i=1
∑T

t=1(yit−1 − ȳi,−1)(γ(yit−1 − ȳi,−1) + (εit − ε̄i))∑n
i=1

∑T
t=1(yit−1 − ȳi,−1)2

plim γ̂fe = γ +
∑n

i=1
∑T

t=1(yit−1 − ȳi,−1)(εit − ε̄i)∑n
i=1

∑T
t=1(yit−1 − ȳi,−1)2

̸= γ

Mas ∑n
i=1

∑T
t=1(yit−1 − ȳi,−1)(εit − ε̄i)/(nT ) é a correlação entre o passado de y e o

erro contemporâneo, que é parte do presente de y. Portanto, esse termo não é zero.

Anderson & Hsiao (1981)

A solução AH é aplicar FD com IV, porque (aqui a questão do balanceamento do
painel importa, por conta da FD)...

∆yit = γ∆yit−1 + β2∆xit + β3∆dit + ∆εit

▶ ∆yit−1 = yit−1 − yit−2 é correlacionado com o termo de erro ∆εit = εit − εit−1,
por conta da contemporaneidade em t− 1.

▶ Mas yit−2 pode ser usado como instrumento para ∆yit−1, já que yit−2 não é
correlacionado com ∆εit, mas é correlacionado com ∆yit−1.

▶ Como há um único instrumento, basta aplicar apropriadamente IV.
▶ O problema é que yit−2 pode ser um instrumento fraco. – i.e., γ pode ser

relativamente pequeno.

xtset i t

xtreg y L.y x d, fe
est sto fe

ivregress 2sls D.y D.x D.d (D.L.y = L2.y), nocon
est sto ah

estout fe ah, cells(b(star fmt(3)) se(par fmt(3)))

--------------------------------------------
fe ah

b/se b/se
--------------------------------------------
L.y 0.583***

(0.003)
LD.y 0.666*

(0.320)
x 0.502***

(0.003)
d 0.771***

(0.021)
D.x 0.520***

(0.081)
D.d 0.786***

(0.123)
_cons -0.655***

(0.046)
--------------------------------------------

Exercício para mostrar que AH é consistente, mas não é eficiente

g gamma_fe = .
g gamma_ha = .
forval j = 1/100 {
set seed `=54321`j''
sort i t
replace u = c+rnormal(0,.2)
replace y = .
bys i: replace y = b1+b2*x+b3*d+u if t == 1
bys i: replace y = gamma*y[_n-1]+b1+b2*x+b3*d+u if t > 1
xtset i t
xtreg y L.y x d, fe
mat fe = e(b)
replace gamma_fe = fe[1,1] in `j'
ivregress 2sls D.y D.x D.d (D.L.y = L2.y), nocon
mat ha = e(b)
replace gamma_ha = ha[1,1] in `j'
}
tw (kdensity gamma_fe) (kdensity gamma_ha), ///
xline(.6, lpattern(dash) lcolor(blue)) xsize(16) ysize(9)

Arellano & Bond (1991)

▶ Segue explorando ∆yit = γ∆yit−1 + β2∆xit + β3∆dit + ∆εit, mas aplicando
como instrumentos yi,t−s para s ≥ 2.

▶ As principais vantagens: utiliza todas as defasagens válidas (yi,t−2, yi,t−3, . . . ),
então há menor risco de instrumentos fracos; e, permite o teste de Sargan.

▶ Se E[yi,t−s∆εit] = 0, pode-se aplicar GMM.
▶ Mas agora há mais instrumentos que parâmetros a se identificar, então é

preciso definir uma função objetivo Q(θ) = ĝ(θ)′Wĝ(θ), em que θ = [γ, β2, β3]
e pode incluir outros eventuais parâmetros auxiliares.
▶ ĝ(θ) = 1

n

∑n
i=1 Z ′

i∆εi(θ) é o vetor de médias amostrais das condições de
momento.

▶ Zi é um vetor contendo yi,t−2, yi,t−3, . . . (e variáveis exógenas).
▶ W é a matriz de pesos.

Exemplo GMM com T = 4

∥Q(θ)∥ = ĝ(θ)′W ĝ(θ)

=

∥∥∥∥∥∥∥∥∥∥


∑n

i=1 yi2∆εi4

n∑n

i=1 yi1∆εi4

n∑n

i=1 yi1∆εi3

n


′ [

w1 0 0
0 w2 0
0 0 w3

]
∑n

i=1 yi2∆εi4

n∑n

i=1 yi1∆εi4

n∑n

i=1 yi1∆εi3

n


∥∥∥∥∥∥∥∥∥∥

=

√
w1

(∑n

i=1 yi2∆εi4

n

)2
+ w2

(∑n

i=1 yi1∆εi4

n

)2
+ w3

(∑n

i=1 yi1∆εi3

n

)2

∆εit = ∆yit − (γ∆yit−1 + β2∆xit + β3∆dit)

θ̂ = argminθ ∥Q(θ)∥

▶ São empreendidos métodos numéricos para computar θ̂ sempre que há “superidentificação”.
▶ Poderíamos ter até 5 instrumentos ao considerar x e d como exógenas.

Arellano & Bover (1995)

A principal inovação é perceber que a equação pode ser tratada em nível:

yit = γyit−1 + β1 + β2xit + β3dit + uit

E explorar como instrumentos múltiplas defasagens das diferenças como
instrumentos ∆yi,t−2, ∆yi,t−3 . . . , já que essas difrenças não seriam correlacionadas
com uit = ci + εit

A vantagem dessa abordagem é que ela não usa diferenças, então:
▶ se ajusta melhor para painéis desbalanceados
▶ permite estudar o efeito de variáveis constantes

clear all
set seed 54321
loc n = 100
loc T = 20
set obs `n'
g i = _n
expand `T'
bys i: gen t = _n
sca gamma = .5
sca beta = .5
g x = rpoisson(9)
sort i t
bys i: g c = rlogistic(.2,.9) if _n == 1
bys i: replace c = c[_n-1] if mi(c)
g u = c+rnormal(0,1.2)
g y = .
bys i: replace y = beta*x+u if t == 1
bys i: replace y = gamma*y[_n-1]+beta*x+u if t > 1
xtset i t

xtreg y L.y x, fe
est sto fe

ivregress 2sls D.y D.x (D.L.y = L2.y), nocon
est sto ah

xtabond y x, nocons lags(1) maxlags(1) pre(x)
est sto ab

gmm (D.y - {gamma}*D.L.y - {beta}*D.x), instruments(L2.y L.x x D.x)
est sto ab2

xtdpdsys y x, lags(1) maxlags(1)
est sto bl

gmm (y - {gamma}*L.y - {beta}*x - {c}), instruments(L.D.y L2.D.y L2.y D.x)
est sto bl2

estout fe ah ab ab2 bl bl2, cells(b(star fmt(3)) se(par fmt(3)))

Ver pág. 40 de https://www.stata.com/manuals13/rgmm.pdf

Comparando FE, AH e AB

g gamma_fe = .
g gamma_ha = .
g gamma_ab = .
forval j = 1/100 {
set seed `=54321`j''
sort i t
replace u = c+rnormal(0,.2)
replace y = .
bys i: replace y = b1+b2*x+b3*d+u if t == 1
bys i: replace y = gamma*y[_n-1]+b1+b2*x+b3*d+u if t > 1
xtset i t
xtreg y L.y x d, fe
mat fe = e(b)
replace gamma_fe = fe[1,1] in `j'
ivregress 2sls D.y D.x D.d (D.L.y = L2.y), nocon
mat ha = e(b)
replace gamma_ha = ha[1,1] in `j'
xtabond y x d, nocons lags(1)
mat ab = e(b)
replace gamma_ab = ab[1,1] in `j'
}
tw (kdensity gamma_fe) (kdensity gamma_ha) (kdensity gamma_ab), ///
xline(.6, lpattern(dash) lcolor(blue)) xsize(16) ysize(9) legend(cols(3))

AB é consistente, e é eficiente

Escolhendo modelos de painel dinâmico: R2, AIC etc. não se aplicam

▶ Estimativas de GMM dependem substancialmente da escolha de Z e W .
▶ Quanto maior a dimensão de Z, mais chance de instrumentos fracos.
▶ 1 etapa (W determinístico) é consistente, mas menos eficiente.
▶ 2 etapas (W estimado) é mais eficiente, mas pode ser inconsistente.

▶ Testes populares:
▶ Sargan verifica se os instrumentos são exógenos.
▶ Arellano-Bond verifica correlação nos resíduos: AR(1) é esperado e AR(2) não

deve ser significativo.
▶ Como se faz na prática:

▶ Evite usar muitos instrumentos, e compare robustez dos coeficientes entre
diferentes especificações.

▶ Use o conhecimento do fenômeno estudado para guiar a escolha de defasagens e
instrumentos.

Exemplo para a equação de Mincer na PNAD-C

cls
clear all
import excel "https://petterini.ufsc.br/files/2024/11/data_fln_filtrada.xlsx", firstrow
g tri = qofd(t)
form tri %tq
replace t = tri
drop tri
*g lnw = ln(w/((30/7)*h))
g lnw = ln(w)
g idade2 = idade^2
global X edu idade idade2 homem branco
reg lnw $X if cond_dom == 1
est sto ols
xtset id t
xtreg lnw edu idade idade2 if cond_dom == 1, fe
est sto fe
ivregress 2sls D.lnw D.edu D.idade D.idade2 (D.L.lnw = L2.lnw) if cond_dom == 1, nocon
est sto ah
xtabond lnw edu idade idade2 if cond_dom == 1, lags(1)
est sto ab1
xtdpdsys lnw edu idade idade2 if cond_dom == 1, lags(1)
est sto ab2
estout ols fe ah ab1 ab2, cells(b(star fmt(3)) se(par fmt(3)))

Exemplo para projeção de taxa de homicídios
clear all
import excel "https://petterini.ufsc.br/files/2024/11/data_projeta_avanca_sc.xlsx", ///
sheet("data") firstrow
loc n = _N
loc N = _N+12
set obs `N'
forvalues i = `n'/`N' {
replace i = "SC" in `i'
replace t = t[_n-1]+1 in `i'

}
fillin i t
drop _fillin
en i, g(id)
xtset id t
g y = tx_hom_total
gl nome_var "Taxa de homicídios"
*xtunitroot fisher y, dfuller lags(1)
ivregress 2sls D.y (D.L.y = L2.y) if i != "SC", nocon vce(cluster i)
sca gamma = e(b)[1,1]
sca sigma = sqrt(e(V)[1,1])

g y_trend = y
g y_minus = y
g y_plus = y
sort id t
bys id: g fe = y-gamma*L.y if t == 2023
bys id: egen fe2 = mean(fe)
bys id: replace fe = fe2
drop fe2
sca z = 1.64485
sum y
sca min = floor(r(min))
sca max = ceil(r(max))
sca interv = ceil((max-min)/5)
bys id: replace y_trend = gamma*y_trend[_n-1]+fe if y_trend == .
bys id: replace y_minus = (gamma-z*sigma)*y_minus[_n-1]+fe if y_minus == .
bys id: replace y_plus = (gamma+z*sigma)*y_plus[_n-1]+fe if y_plus == .

loc regioes "AMAI AMARP AMAUC AMAVI AMEOSC AMERIOS AMESC AMFRI AMMOC AMNOROESTE AMOSC AMPLANORTE
AMPLASC AMREC AMUNESC AMURC AMUREL AMURES AMVALI AMVE GRANFPOLIS SC"

foreach l of loc regioes {
tw (rarea y_minus y_plus t, color(gs12)) ///

(line y_trend t, lcolor(blue) lpattern(solid) lwidth(medium)) ///
(line y_minus t, lcolor(blue) lpattern(dash) lwidth(medium)) ///
(line y_plus t, lcolor(blue) lpattern(dash) lwidth(medium)) if i == "`l'", ///
title("${nome_var} para `l'", justification(center)) ///

xtitle("Anos", angle(90) lcolor(black)) ///
xlabel(2010(1)2035, angle(90) format(%ty)) ///

ylabel(`=min'(`=interv')`=max', grid) yscale(range( )) ///
graphregion(color(gs15)) ///

xsize(16) ysize(9) legend(off) ///
xline(2023, lcolor(red) lpattern(dash) lwidth(medium)) ///

text(20 2015 "Brasil", place(n)) ///
yline(19, lcolor(red) lpattern(dash) lwidth(medium)) ///
text(6 2015 "Itália", place(n)) ///
yline(5 2015, lcolor(red) lpattern(dash) lwidth(medium))

graph export "C:\Users\73771430020\Desktop\`l'.png", replace
}

Exemplo para projeção de vendas de motos

cls
clear all
import excel "https://petterini.ufsc.br/files/2024/11/data_motos.xlsx", ///
sheet("data") firstrow
local n = _N
local N = _N + 12
set obs `N'
forvalues i = `=`n' + 1' / `N' {
replace i = i[_n-1] in `i'
replace t = t[_n-1]+1 in `i'

}
fillin i t
drop _fillin
sort i t
replace ano = 2013 if mi(ano)
bys i: replace mes = _n-84 if mi(mes)
replace modelo = modelo[_n-1] if mi(modelo)
bys i: replace p = p[_n-1] if mi(p)

gen cc = .
replace cc = 150 if modelo == "CG150"
replace cc = 125 if modelo == "BIZ"
replace cc = 125 if modelo == "CG125"
replace cc = 250 if modelo == "COMET"
replace cc = 600 if modelo == "HORNET"
replace cc = 250 if modelo == "MIRAGE"
replace cc = 115 if modelo == "NEO"
replace cc = 150 if modelo == "NXR"
replace cc = 750 if modelo == "SHADOW"
replace cc = 660 if modelo == "XT"
replace cc = 125 if modelo == "XTZ"
replace cc = 125 if modelo == "YBR"
g flex = 0
replace flex = 1 if modelo == "CG150" & ano >= 2009
replace flex = 1 if modelo == "BIZ" & ano >= 2011
replace flex = 1 if modelo == "NXR" & ano >= 2010
replace p = p/1000
sort t
bys t: egen qt = sum(q)
g s = q/qt if q != .
drop qt
g lns = ln(s)
sort i t

xtset i t
xtunitroot fisher lns, dfuller lags(1)
reg lns p i.cc flex i.mes i.ano
est sto ols
xtreg lns p flex i.mes i.ano, fe
est sto fe
xi: xtdpdsys lns p i.cc flex i.mes i.ano, lags(1)
est sto ab
xi: xtdpdsys lns L.lns p i.cc flex i.mes i.ano
est sto ab2
xi: xtdpdsys lns L12.lns p i.cc flex i.mes i.ano
est sto ab12
estat abond
estout ols fe ab ab2 ab12, cells(b(star fmt(3)) se(par fmt(3)))

forecast create projecoes
forecast estimates ab12
forecast solve, prefix(proj_) begin(85) end(96)

g proj_s = exp(proj_lns)

loc motos "CG150 BIZ CG125 COMET HORNET MIRAGE NEO NXR SHADOW XT XTZ YBR"
loc codes 1 2 3 4 5 6 7 8 9 10 11 12

forvalues j = 1/12 {
local l : word `j' of `motos'
local code : word `j' of `codes'
line proj_s t if i == `code', lcolor(blue) lpattern(dash) lwidth(medium) ///
title("Projeção para `l'", justification(center)) ///
xtitle("t", angle(90) lcolor(black)) ///

xlabel(1(3)96, angle(90) format(%ty)) ///
graphregion(color(gs15)) ///
xsize(16) ysize(9) legend(off) ///
xline(85, lcolor(red) lpattern(dash) lwidth(medium))

graph export "C:/Users/73771430020/Desktop/`l'.png", replace
}

Exemplo de projeção da varição do preço do querosene de aviação
clear all
import excel "https://petterini.ufsc.br/files/2024/11/querosene.xlsx", ///
sheet("data") firstrow
/*
loc regioes "NO NE CO SU SE BR"
foreach l of loc regioes {
kdensity p if i == "`l'"

}
*/
loc n = _N
loc N = _N+10
set obs `N'
forvalues i = `n'/`N' {
replace i = "BR" in `i'
replace t = t[_n-1]+1 in `i'

}
fillin i t
drop _fillin
en i, g(id)
xtset id t
*xtline p
g y = ln(p/L.p)
xtunitroot fisher y, dfuller lags(1)
replace date1 = date1[_n-1]+7 if mi(date1)
ivregress 2sls D.y (D.L.y = L2.y) if i != "BR", nocon vce(cluster id)
g gamma = e(b)[1,1]
sort id
forv l = 1/6 {
bys id: g sigma`l' = sqrt(abs(y[70]-y[69])*e(V)[1,1]) /// método delta
}
sort id t

bys id: g fe = y-gamma*L.y if t == 70
bys id: egen fe2 = mean(fe)
bys id: replace fe = fe2
drop fe2
sca z = 1.96
sum y
g y_trend = y
g y_minus = y
g y_plus = y
sort id
forv l = 1/6 {
replace y_trend = gamma*y_trend[_n-1]+fe if y_trend == . & id == `l'
replace y_minus = (gamma-z*sigma`l')*y_trend[_n-1]+fe if y_minus == . & id == `l'
replace y_plus = (gamma+z*sigma`l')*y_trend[_n-1]+fe if y_plus == . & id == `l'
}
loc regioes "NO NE CO SU SE BR"
foreach l of loc regioes {
tw (rarea y_minus y_plus t, color(gs12)) ///

(line y_trend t, lcolor(blue) lpattern(solid) lwidth(medium)) ///
(line y_minus t, lcolor(blue) lpattern(dash) lwidth(medium)) ///
(line y_plus t, lcolor(blue) lpattern(dash) lwidth(medium)) if i == "`l'", ///
title("variação do preço do querosene de aviação no `l'", justification(center)) ///
xtitle("semanas", angle(90) lcolor(black)) ///
xlabel(0(5)80, angle(90)) ///
graphregion(color(gs15)) ///
xsize(16) ysize(9) legend(off) ///
xline(70, lcolor(red) lpattern(dash) lwidth(medium)) ///
yline(0 80, lcolor(red) lpattern(dash) lwidth(medium))
graph export "C:\Users\73771430020\Desktop\g`l'.png", replace

}

y ∈ {0, 1}

▶ y = 1 se y∗ ≥ 0 e y = 0 se y∗ < 0



y∗
11 = γy10 + β1 + β2x11 + β3d11 + c1 + ε11

y∗
12 = γy11 + β1 + β2x12 + β3d12 + c1 + ε12

...
y∗

1T = γy1(T −1) + β1 + β2x1T + β3d1T + c1 + ε1T

...
y∗

n1 = γyn0 + β1 + β2xn1 + β3dn1 + cn + εn1

y∗
n2 = γyn1 + β1 + β2xn2 + β3dn2 + cn + εn2

...
y∗

nT = γyn(T −1) + β1 + β2xnT + β3dnT + cn + εnT

Breve revisão da alfabetização

Probit e Logit se γ = 0 e T = 1... e um único x

Pr(y = 1 | x̄, d = 1)− Pr(y = 1 | x̄, d = 0)

Pr(y∗ > 0 | x̄, d = 1)− Pr(y∗ > 0 | x̄, d = 0)

Pr(β1 + β2x̄ + β3 + u > 0)− Pr(β1 + β2x̄ + u > 0) = .25

β1 = 0 e x̄ = 0 e u ∼ N(0, 1)

Pr(u > −β3)− Pr(u > 0) = .25

Φ(β3)− Φ(0) = .25⇒ Φ(β3) = .75⇒ β3 = Φ−1(.75) ≈ .67448975

cls
clear all
mat results = J(30,6,.)
forv i=1(1)30 {
clear
set obs `=1000*`i''
set seed 54321`i'
g x = runiform()-.5
g d = runiform() > .5
g y = x+invnormal(.75)*d+rnormal() > 0
egen x_mean = mean(x)
probit y x d
sca delta_probit = normal(_b[_cons]+_b[x]*x_mean+_b[d])-normal(_b[_cons]+_b[x]*x_mean)
mat results[`i',1] = delta_probit
mat results[`i',3] = _b[x]
mat results[`i',5] = _b[d]
logit y x d
sca delta_logit = invlogit(_b[_cons]+_b[x]*x_mean+_b[d])-invlogit(_b[_cons]+_b[x]*x_mean)
mat results[`i',2] = delta_probit
mat results[`i',4] = _b[x]
mat results[`i',6] = _b[d]

}
mat l results

Principias elementos de ML

P (yi = 1 | xxxi) = F (β′xxxi)⇒ P ({(yi,xxxi)}ni=1) i.i.d.=
n∏

i=1

[
F (β′xxxi)

]yi
[
1− F (β′xxxi)

]1−yi

ln L(β) =
n∑

i=1

[
yi ln F (β′xxxi)+(1−yi) ln

(
1−F (β′xxxi)

)]
⇒
{
∇(β) = ∑n

i=1 (yi − F (β′xxxi))xxxi

H(β) = −∑n
i=1 f(β′xxxi)xxx′

ixxxi

β̂j+1 ← β̂j − [H(β̂j)]−1∇(β̂j)⇒ β̂
d−→ N

(
000,
(
−H(β̂)

)−1
)

Vâr(β̂) =
(
−H(β̂)

)−1
⇒ se(β̂k) =

√[
Vâr(β̂)

]
kk
⇒ zk = β̂k

se(β̂k)

Pseudo-R2 = 1− ln L(β̂)/ ln L(β̂0)

cls
clear all

set obs 100
set seed 54321
g x = runiform()
g y = x+rnormal() > 0

probit y x
estat ml

program drop _all
program myprobit
args lnf xb
qui{
replace `lnf' = $ML_y1*ln(normal(`xb'))+(1-$ML_y1)*ln(normal(-`xb'))

}
end
ml model lf myprobit (y = x)
ml check
ml maximize
estat ml

estout pro ml, cells(b(star fmt(3)) se(par fmt(3)))

mat b = (0 \ 0)
sca tol = 1e-5
sca diff = 1
sca iter = 0
sca max_iter = 100
g xb = .
g p = .
g score = .
g g0 = .
g g1 = .
g h = .
g h2 = .
g denom = .
g H00_part = .
g H01_part = .
g H11_part = .
g log_likelihood_contrib = .
while (diff > tol) & (iter < max_iter) {
replace xb = b[1,1]+b[2,1]*x
replace p = normal(xb)
replace score = y-p
replace g0 = score
replace g1 = score*x
replace h = normalden(xb)
replace h2 = h*h
replace denom = p*(1-p)
replace H00_part = h2/denom
replace H01_part = (h2*x)/denom
replace H11_part = (h2*(x^2))/denom

sum H00_part
sca H00 = -r(sum)
sum H01_part
sca H01 = -r(sum)
sum H11_part
sca H11 = -r(sum)
sum g0
sca G0 = r(sum)
sum g1
sca G1 = r(sum)
mat G = ( G0 \ G1 )
mat H = ( H00, H01 \ H01, H11 )
mat H_inv = inv(H)
mat delta = -H_inv*G
mat b = b+delta
sca diff = sqrt(delta[1,1]^2 + delta[2,1]^2)
replace log_likelihood_contrib = y*log(p)+(1-y)*log(1-p)
sum log_likelihood_contrib
sca log_likelihood = r(sum)
di "Iteration " iter
di ": b0 = " b[1,1] ", b1 = " b[2,1]
di ", diff = " diff
di ", log-likelihood = " log_likelihood
sca iter = iter + 1
}

C-Logit: Logit com efeitos fixos para γ = 0 e T > 1
P (yit = 1 | xxxit, ci) =

exp(β′xxxit + ci)
1 + exp(β′xxxit + ci)

T = 2 temos yyyi ∈ {(0, 0), (0, 1), (1, 0), (1, 1)}

P (yyyi = (0, 0)) =
1

1 + exp(β′xxx1t + ci)
×

1
1 + exp(β′xxx2t + ci)

P (yyyi = (1, 0)) =
exp(β′xxxi1 + ci)

1 + exp(β′xxxi1 + ci)
×

1
1 + exp(β′xxxi2 + ci)

= exp(ci)
exp(β′xxxi1)

1 + exp(β′xxxi1 + ci)
×

1
1 + exp(β′xxxi2 + ci)

P (yyyi = (0, 1)) =
1

1 + exp(β′xxxi1 + ci)
×

exp(β′xxxi2 + ci)
1 + exp(β′xxxi2 + ci)

= exp(ci)
1

1 + exp(β′xxxi1 + ci)
×

exp(β′xxxi2)
1 + exp(β′xxxi2 + ci)

P (yyyi = (1, 1)) =
exp(β′xxxi1 + ci)

1 + exp(β′xxxi1 + ci)
×

exp(β′xxxi2 + ci)
1 + exp(β′xxxi2 + ci)

= exp(2ci)
exp(β′xxxi1)

1 + exp(β′xxxi1 + ci)
×

exp(β′xxxi2)
1 + exp(β′xxxi2 + ci)

P (yyyi | xxxit, ci) =
exp(yi1β′xxxi1 + yi1ci)
1 + exp(β′xxxi1 + ci)

×
exp(yi2β′xxxi2 + yi2ci)
1 + exp(β′xxxi2 + ci)

= exp(ci

T∑
t=1

yit))
T∏

t=1

exp(yitβ′xxxit)
1 + exp(β′xxxit + ci)

P (yi1 + yi2 = 0 | xxxit, ci) = P (yyyi = (0, 0) | xxxit, ci)

P (yi1 + yi2 = 1 | xxxit, ci) = P (yyyi = (1, 0) | xxxit, ci) + P (yyyi = (0, 1) | xxxit, ci)

P (yi1 + yi2 = 2 | xxxit, ci) = P (yyyi = (1, 1) | xxxit, ci)

P

( T∑
t=1

yit = s | xxxi, ci

)
=
∑

s

exp(ci

T∑
t=1

yit))
T∏

t=1

exp(yitβ′xxxit)
1 + exp(β′xxxit + ci)

P (yyyi |
T∑

t=1

yit = s,xxxit) =

∏T

t=1 exp(ci

∑T

t=1 yit))
exp(yitβ′xxxit)

1 + exp(β′xxxit + ci)∑
s

∏T

t=1 exp(ci

∑T

t=1 yit))
exp(yitβ′xxxit)

1 + exp(β′xxxit + ci)

=
exp(yitβ′xxxit)∑

i é do tipo s
exp(yitβ′xxxit)

clear all
set seed 54321
loc n = 1000
loc T = 10
set obs `n'
g i = _n
expand `T'
bys i: g t = _n
g x = 5*(runiform()-.5)
g d = runiform()>.5
sort i t
bys i: g c = 5*(runiform()-.5) if _n == 1
bys i: replace c = c[_n-1] if mi(c)
g y = x+d+c+rlogistic() > 0
sum y
logit y x d c
est sto logit_com_c
logit y x d
est sto logit_sem_c
xtset i t
xtlogit y x d, fe
est sto xtlogit
clogit y x d, gr(i)
est sto clogit
estout logit_com_c logit_sem_c xtlogit clogit, cells(b(star fmt(3)) se(par fmt(3)))

clogit y x d, gr(i)
mat b = e(b)

range xval -2.5 2.5 11
g p_d1 = invlogit(b[1,1]*xval+b[1,2])
g p_d0 = invlogit(b[1,1]*xval)

tw (line p_d1 xval, lcolor(blue) lpattern(solid) lwidth(medium)) ///
(line p_d0 xval, lcolor(red) lpattern(dash) lwidth(medium)), ///
xlabel(-2.5(.5)2.5, format(%2.1f)) ylabel(0(.2)1, format(%2.1f)) ///
legend(label(1 "P(y=1|x,d=1)") label(2 "P(y=1|x,d=0)") ///
placement(bottom) col(2) region(margin(0) lstyle(none))) graphregion(color(white))

Método Delta

▶ Deseja-se o erro-padrão de uma função diferenciável g(β̂), talque g : R→ R.

g(β̂) ≈ g(β) + g′(β)(β̂ − β)⇒ Vâr(g(β̂)) ≈ [g′(β̂)]2Vâr(β̂)

▶ Se β̂ = (β̂1, β̂2, . . . , β̂K)′ e g : RK → R, o gradiente é:

∇g(β) =
(

∂g

∂β1
,

∂g

∂β2
, . . . ,

∂g

∂βK

)′

Vâr(g(β̂)) ≈ ∇g(β̂)′Vâr(β̂)∇g(β̂)

Pr(y = 1 | x, d = 1) = g(β) = Λ(β1x+β2)⇒ ∇g(β̂) =
[
Λ(β̂1 + β̂2)(1− Λ(β̂1 + β̂2))x
Λ(β̂1 + β̂2)(1− Λ(β̂1 + β̂2))

]

Pr(y = 1 | x, d = 0) = g(β) = Λ(β1x)⇒ ∇g(β̂) = Λ(β̂1)(1− Λ(β̂1))x

Vâr(β̂) =
(
−H(β̂)

)−1

drop p_d*
clogit y x d, group(i)
mat b = e(b)
mat V = e(V)
g p_d1 = .
g s_d1 = .
g l_d1 = .
g u_d1 = .
g p_d0 = .
g s_d0 = .
g l_d0 = .
g u_d0 = .

forv j = 1/11 {
replace p_d1 = invlogit(b[1,1]*xval[`j']+b[1,2]) in `j'
mat grad_d1 = (invlogit(b[1,1]*xval[`j']+b[1,2])*(1-invlogit(b[1,1]*xval[`j']+b[1,2]))*xval[`j'] \ ///

invlogit(b[1,1]*xval[`j']+b[1,2])*(1-invlogit(b[1,1]*xval[`j']+b[1,2])))
mat var_d1 = grad_d1'*V*grad_d1
replace s_d1 = sqrt(var_d1[1,1]) in `j'
replace l_d1 = p_d1-1.96*s_d1 in `j'
replace u_d1 = p_d1+1.96*s_d1 in `j'
replace p_d0 = invlogit(b[1,1]*xval[`j']) in `j'
sca grad_d0 = invlogit(b[1,1]*xval[`j'])*(1-invlogit(b[1,1]*xval[`j']))*xval[`j']
sca var_d0 = (grad_d0^2)*V[1,1]
replace s_d0 = sqrt(var_d0) in `j'
replace l_d0 = p_d0-1.96*s_d0 in `j'
replace u_d0 = p_d0+1.96*s_d0 in `j'

}

tw (rarea l_d1 u_d1 xval, color(ltblue%50) lcolor(blue)) (line p_d1 xval, lcolor(blue) lpattern(solid) lwidth(medium)) ///
(rarea l_d0 u_d0 xval, color(ltred%25) lcolor(red)) (line p_d0 xval, lcolor(red) lpattern(dash) lwidth(medium)), ///
xlabel(-2.5(.5)2.5, format(%2.1f)) ylabel(0(.2)1, format(%2.1f)) ///
legend(label(1 "P(y=1|x,d=1)") label(2 "P(y=1|x,d=0)") ///
placement(bottom) col(2) region(margin(0) lstyle(none))) graphregion(color(white))

drop p_d*

g p_d1 = .
g l_d1 = .
g u_d1 = .
g p_d0 = .
g l_d0 = .
g u_d0 = .

forv j = 1/11 {
clogit y x d, group(i)
nlcom p1: invlogit(_b[x]*xval[`j']+_b[d]), post
replace p_d1 = _b[p1] in `j'
replace l_d1 = _b[p1]-1.96*_se[p1] in `j'
replace u_d1 = _b[p1]+1.96*_se[p1] in `j'
clogit y x d, group(i)
nlcom p0: invlogit(_b[x]*xval[`j']), post
replace p_d0 = _b[p0] in `j'
replace l_d0 = _b[p0]-1.96*_se[p0] in `j'
replace u_d0 = _b[p0]+1.96*_se[p0] in `j'

}

tw (rarea l_d1 u_d1 xval, color(ltblue%50) lcolor(blue)) (line p_d1 xval, lcolor(blue) lpattern(solid) lwidth(medium)) ///
(rarea l_d0 u_d0 xval, color(ltred%25) lcolor(red)) (line p_d0 xval, lcolor(red) lpattern(dash) lwidth(medium)), ///
xlabel(-2.5(.5)2.5, format(%2.1f)) ylabel(0(.2)1, format(%2.1f)) ///
legend(label(1 "P(y=1|x,d=1)") label(2 "P(y=1|x,d=0)") ///
placement(bottom) col(2) region(margin(0) lstyle(none))) graphregion(color(white))

union.dta

cls
clear all
webuse union
describe
sum

▶ year e idcode
▶ age: anos de idade do trabalhador
▶ grade: anos de escolaridade completados
▶ not_smsa: = 1 se reside fora de uma área metropolitana
▶ south: = 1 vive na região sul dos EUA
▶ union: = 1 afiliação sindical
▶ black: = 1 negro

gl X age black grade not_smsa south

logit union $X
est sto logit
xtset idcode year
xtlogit union $X, fe
est sto xtlogit
estout logit xtlogit, cells(b(star fmt(3)) se(par fmt(3)))

tab grade

sum age
sca age_m = r(mean)
sum not_smsa
sca not_smsa_m = r(mean)

range edu 0 18 19

g p_d1 = .
g l_d1 = .
g u_d1 = .
g p_d0 = .
g l_d0 = .
g u_d0 = .

130 / 221



forv j = 1/19 {
clogit union age grade not_smsa south, group(idcode)
nlcom p1: invlogit(_b[age]*age_m+_b[grade]*edu[`j']+_b[not_smsa]*not_smsa_m+_b[south]), post
replace p_d1 = _b[p1] in `j'
replace l_d1 = _b[p1]-1.96*_se[p1] in `j'
replace u_d1 = _b[p1]+1.96*_se[p1] in `j'
clogit union age black grade not_smsa south, group(idcode)
nlcom p0: invlogit(_b[age]*age_m+_b[grade]*edu[`j']+_b[not_smsa]*not_smsa_m), post
replace p_d0 = _b[p0] in `j'
replace l_d0 = _b[p0]-1.96*_se[p0] in `j'
replace u_d0 = _b[p0]+1.96*_se[p0] in `j'

}

tw (rarea l_d1 u_d1 edu, color(ltblue%50) lcolor(blue)) (line p_d1 edu, lcolor(blue) lpattern(solid)) ///
(rarea l_d0 u_d0 edu, color(ltred%25) lcolor(red)) (line p_d0 edu, lcolor(red) lpattern(dash)), ///
xlabel(0(1)18, format(%2.1f)) ylabel(0(.2)1, format(%2.1f)) ///
legend(label(1 "P(y=1 | sul)") label(2 "P(y=1 | não sul)") ///
placement(bottom) col(2) region(margin(0) lstyle(none))) graphregion(color(white))

Acrescentando a dinâmica

Probit com |γ| < 1 e o problema da condição inicial


y∗
11 = γ y10 + βx11 + c1 + ε11

y∗
12 = γ y11 + βx12 + c1 + ε12

y∗
13 = γ y12 + βx13 + c1 + ε13

y∗
21 = γ y20 + βx21 + c2 + ε21

y∗
22 = γ y21 + βx22 + c2 + ε22

y∗
23 = γ y22 + βx23 + c2 + ε23

;

εi1
εi2
εi3

 | (xit, ci) ∼ N


0

0
0

 ,

 1 ρ12 ρ13
ρ12 1 ρ23
ρ13 ρ23 1




ε ∼ N (0, Σ) ∴ Σ =

 1 ρ12 ρ13
ρ12 1 ρ23
ρ13 ρ23 1

⇒ f(ε) = 1
(2π)3/2|Σ|1/2 exp

(
−1

2εT Σ−1ε

)

Pr(y1 = 0, y2 = 0, y3 = 0) = Pr(ε1 ≤ 0, ε2 ≤ 0, ε3 ≤ 0) =

∫ 0

−∞

∫ 0

−∞

∫ 0

−∞

f(ε1, ε2, ε3) dε1 dε2 dε3,

Pr(y1 = 0, y2 = 0, y3 = 1) = Pr(ε1 ≤ 0, ε2 ≤ 0, ε3 > 0) =

∫ 0

−∞

∫ 0

−∞

∫ ∞

0

f(ε1, ε2, ε3) dε1 dε2 dε3,

Pr(y1 = 0, y2 = 1, y3 = 0) = Pr(ε1 ≤ 0, ε2 > 0, ε3 ≤ 0) =

∫ 0

−∞

∫ ∞

0

∫ 0

−∞

f(ε1, ε2, ε3) dε1 dε2 dε3,

Pr(y1 = 0, y2 = 1, y3 = 1) = Pr(ε1 ≤ 0, ε2 > 0, ε3 > 0) =

∫ 0

−∞

∫ ∞

0

∫ ∞

0

f(ε1, ε2, ε3) dε1 dε2 dε3,

Pr(y1 = 1, y2 = 0, y3 = 0) = Pr(ε1 > 0, ε2 ≤ 0, ε3 ≤ 0) =

∫ ∞

0

∫ 0

−∞

∫ 0

−∞

f(ε1, ε2, ε3) dε1 dε2 dε3,

Pr(y1 = 1, y2 = 0, y3 = 1) = Pr(ε1 > 0, ε2 ≤ 0, ε3 > 0) =

∫ ∞

0

∫ 0

−∞

∫ ∞

0

f(ε1, ε2, ε3) dε1 dε2 dε3,

Pr(y1 = 1, y2 = 1, y3 = 0) = Pr(ε1 > 0, ε2 > 0, ε3 ≤ 0) =

∫ ∞

0

∫ ∞

0

∫ 0

−∞

f(ε1, ε2, ε3) dε1 dε2 dε3,

Pr(y1 = 1, y2 = 1, y3 = 1) = Pr(ε1 > 0, ε2 > 0, ε3 > 0) =

∫ ∞

0

∫ ∞

0

∫ ∞

0

f(ε1, ε2, ε3) dε1 dε2 dε3.

Soluções mais populares

▶ FIML é complicado computacionalmente:
P ({yyyi}Ti=1 | todos yi0, xit) =

∫ ∏n
i=1

[ ∫
yyyi

f(εεεi(parâmetros))dεεε
]
f(c)dc

▶ Heckman (1978) (Orme (2001) melhora) trata parametricamente y0, e mantém
c como um efeito aleatório, mas ainda é complicado computacionalmente:
P ({yyyi}Ti=1 | todos yi0, xit) =

∫ ∏n
i=1 Pr(y1i, ..., yiT | controles de y0i)f(c)dc

▶ Wooldridge (2005) modela c como função de y0 (e outras variáveis iniciais), e
monta uma função de verossimilhança condicionada em y0. Isso contorna
problemas computacionais (e dá aplicabilidade para outros modelos).

ci = cte + ρy0 + controles + ηi ; ηi é um componente aleatório

P ({yyyi}Ti=1 | todos yi0, xit) =
∫ n∏

i=1
Pr(y1i, ..., yiT | controles de ci)f(η)dη

Simulando um Probit-Wooldridge

clear all
set seed 54321
loc n = 100
loc T = 10
loc reps = 50
mat m_probit = J(`reps', 2, .)
mat m_wooldr = J(`reps', 2, .)
tic
forv rep = 1/`reps' {
clear
set obs `n'
g i = _n
expand `T'
bys i: g t = _n
sca gamma = .25
sca beta = 1
sort i t
bys i: gen c = rlogistic(0,2) if _n == 1
bys i: replace c = c[_n-1] if mi(c)
g x = rpoisson(9)
g u = c+rnormal()
g y = runiform() > .5 if t == 1
bys i: replace y = gamma*y[_n-1]+beta*x+u > 0 if t > 1

xtset i t
probit y L.y x, iter(30)
mat m_probit[`rep',1] = _b[L.y]
mat m_probit[`rep',2] = _b[x]
// Wooldridge
bys i: g y0 = y[1]
bys i: egen x_mean = mean(x)
xtprobit y L.y x y0 x_mean if t > 1, re i(i) iter(30)
mat m_wooldr[`rep',1] = _b[L.y]
mat m_wooldr[`rep',2] = _b[x]
}

svmat m_probit, names(p)
svmat m_wooldr, names(w)

sum p* w*

toc

Exemplo com a PNAD-C
cls
clear all
import excel "https://petterini.ufsc.br/files/2024/11/data_fln_filtrada.xlsx", firstrow
g tri = qofd(t)
form tri %tq
replace t = tri
drop tri
keep if pea == 1 & cond_dom == 1
gl X idade homem branco edu
probit ocup $X
est sto probit_n
xtset id t
probit ocup L.ocup $X
est sto probit_l
sort id t
bys id: g ocup0 = ocup[1]
bys id: egen idade_mean = mean(idade)
bys id: egen edu_mean = mean(edu)
xtprobit ocup L.ocup $X ocup0 idade_mean edu_mean if t > 1, ///
re i(id) iter(30)
est sto probit_w
estout probit_n probit_l probit_w, cells(b(star fmt(3)) se(par fmt(3)))
disp normal(_b[L.ocup])
disp normal(_b[L.ocup])-.5

Keane & Wolpin (1997)

Resumo: 11 anos de observações de uma amostra de jovens do coorte de 1979 das
National Longitudinal Surveys of Labor Market Experience (NLSY).
cls
clear all
*ssc install bcuse
*http://fmwww.bc.edu/ec-p/data/wooldridge/datasets.list.html
bcuse keane
keep if black == 1 & numyrs == 11
xtset id year
probit employ L.employ
est sto probit_normal
bys id: g employ0 = employ[1]
xtprobit employ L.employ employ0 if year > 81, re i(id) iter(30)
est sto probit_wooldridge
estout probit_normal probit_wooldridge, cells(b(star fmt(3)) se(par fmt(3)))

y ≥ mínimo

▶ y = y∗ se y∗ > 0 e y = 0 se y∗ ≤ 0 no caso contrário (abordagem de censura)



y∗
11 = γy10 + β1 + β2x11 + β3d11 + c1 + ε11

y∗
12 = γy11 + β1 + β2x12 + β3d12 + c1 + ε12

...
y∗

1T = γy1(T −1) + β1 + β2x1T + β3d1T + c1 + ε1T

...
y∗

n1 = γyn0 + β1 + β2xn1 + β3dn1 + cn + εn1

y∗
n2 = γyn1 + β1 + β2xn2 + β3dn2 + cn + εn2

...
y∗

nT = γyn(T −1) + β1 + β2xnT + β3dnT + cn + εnT

A verossimilhança padrão

ln L(β) =
n∑

i=1

[
I(yi > 0) ln f(yi − β′xxxi) + I(yi = 0) ln F (−β′xxxi)

]

Pr(yi = 0) = Pr(y∗
i ≤ 0) = Pr(β′xxxi + εi ≤ 0) = Pr(εi ≤ −β′xxxi) = F (−β′xxxi)

βj+1 ← βj − [H(βj)]−1∇(βj)

Tobit

y =
{

y∗ se y∗ > 0,

0 se y∗ ≤ 0,
; y∗ = β′xxxi + ε ; ε ∼ N(0, σ2)

L(β, σ) =
n∏

i=1

[
1√

2πσ2
exp

(
−(yi − β′xxxi)2

2σ2

)]I(yi>0)

·
[
Φ
(
−β′xxxi

σ

)]1−I(yi>0)

L(β, σ, σc) =
n∏

i=1

∫ ∞

−∞

T∏
t=1

[
1√

2πσ2
exp

(
−(yit − β′xxxi − ci)2

2σ2

)]I(yit>0)

·
[
Φ
(
−β′xxxi + ci

σ

)]1−I(yit>0)
f(ci) dci

143 / 221



Tobit-Wooldridge
clear all
set seed 54321
loc n = 100
loc T = 10
loc reps = 50
mat m_tobit = J(`reps', 2, .)
mat m_wooldr = J(`reps', 2, .)
tic
forv rep = 1/`reps' {
clear
set obs `n'
g i = _n
expand `T'
bys i: g t = _n
sca gamma = .25
sca beta = 1
sort i t
bys i: gen c = rlogistic(0,2) if _n == 1
bys i: replace c = c[_n-1] if mi(c)
g x = rpoisson(9)
g u = c+rnormal()
g y = max(0, beta*x+u) if t == 1
g y_star = .
forv j = 2/`T' {
bys i: replace y_star = gamma*y[_n-1]+beta*x+u if t == `j'
bys i: replace y = max(0,y_star) if t == `j'

}

xtset i t
tobit y L.y x, ll(0) iter(30)
mat m_tobit[`rep',1] = _b[L.y]
mat m_tobit[`rep',2] = _b[x]
// Wooldridge
bys i: gen y0 = y[1]
bys i: egen x_mean = mean(x)
xttobit y L.y x y0 x_mean if t > 1, re ll(0) i(i) iter(30)
mat m_wooldr[`rep',1] = _b[L.y]
mat m_wooldr[`rep',2] = _b[x]
}
toc
svmat m_tobit, names(to)
svmat m_wooldr, names(w)
sum to* w*

Exemplo com taxa de homicídios municipal no Tobit

clear all
import excel "https://petterini.ufsc.br/files/2024/11/data_tx_hom_mun_sc.xlsx", sheet("data") firstrow
en municipio, g(i)
xtset i ano
reg tx L.tx, vce(cluster i)
est sto ols
reg tx L.tx if tx != 0, vce(cluster i)
est sto ols2
ivregress 2sls D.tx (D.L.tx = L2.tx), nocon vce(cluster i)
est sto ah
ivregress 2sls D.tx (D.L.tx = L2.tx) if tx != 0, nocon vce(cluster i)
est sto ah2
tobit tx L.tx, ll(0) vce(cluster i)
est sto to
bys i: gen tx0 = tx[1]
xttobit tx L.tx tx0 if ano > 2010, re ll(0) i(i) iter(30)
est sto wo
estout ols ols2 ah ah2 to wo, cells(b(star fmt(3)) se(par fmt(3)))

mínimo ≤ y ≤ máximo

Survival (censura na direita)

O principal caso envolve y ≥ 0 representando duração de um evento.

y =
{

y∗ se y∗ ≤ τi,

τi se y∗ > τi,

L(β, σ) =
n∏

i=1

[
f(yi − β′xxxi)

]I(yi<τi) ·
[
1− F

(
τi − β′xxxi

)]1−I(yi<τi)

L(β, σ, σc) =
n∏

i=1

∫ ∞

−∞

T∏
t=1

[
f(yi − β′xxxi − ci)

]I(yi<τi)

·
[
1− F

(
τi − β′xxxi − ci

)]1−I(yi<τi) f(ci) dci

LogNormal
Sob a parametrização µ = β′xxx + c e σ > 0 é o desvio padrão do logaritmo:

▶ fdp

f(t) = 1
t · σ
√

2π
· exp

(
−(ln t− (β′xxx + c))2

2σ2

)
▶ cdf

F (t) = Φ
( ln t− (β′xxx + c)

σ

)
▶ esperança

E(t) = exp
(

β′xxx + c + σ2

2

)
▶ função quantílica

t = exp
(
µ + σ · Φ−1(U)

)
, U ∼ Uniforme(0, 1)

Weibull
Sob a parametrização λ = exp(−α(β′xxx + c)), onde α > 0 é o parâmetro de forma:

▶ fdp

f(t) = exp(−α(β′xxx + c)) · α · tα−1 · exp
(
− exp(−α(β′xxx + c)) · tα)

▶ cdf
F (t) = 1− exp

(
− exp(−α(β′xxx + c)) · tα)

▶ esperança
E(t) = exp

(
β′xxx + c

)
· Γ
(

1 + 1
α

)
▶ função quantílica

t =
(
− ln U

exp(−α(β′xxx + c))

)1/α

Exemplo “Table 22.1” do Wooldridge (2010)
cls
clear all
bcuse recid
* Definindo variáveis
generate f = 1 - cens
stset durat, fail(f)
global X workprg priors tserved felon alcohol drugs black married educ age
* Inicializando matriz para guardar os resultados
matrix AIC_BIC = J(4, 2, .)
matrix colnames AIC_BIC = AIC BIC
matrix rownames AIC_BIC = Weibull Lognormal Weibull_ML Lognormal_ML
* Modelo Weibull padrão
streg $X, dist(weibull) nohr difficult
est store weib
estat ic
matrix results = r(S)
matrix AIC_BIC[1, 1] = results[1, 5] // AIC
matrix AIC_BIC[1, 2] = results[1, 6] // BIC
stcurve, survival title("Survival Curve (Weibull)") ///

xlabel(0(10)50) ylabel(0(0.2)1) ///
saving(weibull_survival, replace)

* Modelo Lognormal padrão
streg $X, dist(lognormal) difficult
est store lognormal
estat ic
matrix results = r(S)
matrix AIC_BIC[2, 1] = results[1, 5] // AIC
matrix AIC_BIC[2, 2] = results[1, 6] // BIC 150 / 221



* Weibull ajustado manualmente com ML
capture program drop my_weibull
program define my_weibull

args lnf xb a
tempvar alph
quietly {

loc t "$ML_y1"
gen double `alph' = exp(`a')
replace `lnf' = `xb' + `a' + (`alph' - 1) * ln(`t') - exp(`xb') * (`t'^`alph')
replace `lnf' = -exp(`xb') * (`t'^`alph') if cens == 1

}
end
ml model lf my_weibull (durat = $X) ()
ml check
ml max, difficult iterate(50)
est store weib2
estat ic
matrix results = r(S)
matrix AIC_BIC[3, 1] = results[1, 5] // AIC
matrix AIC_BIC[3, 2] = results[1, 6] // BIC

* Lognormal ajustado manualmente com ML
capture program drop my_lognormal
program define my_lognormal

args lnf xb sigma
tempvar sigmasq
quietly {

loc t "$ML_y1"
gen double `sigmasq' = exp(2 * `sigma')
replace `lnf' = -ln(`t') - ln(sqrt(2*_pi)) - ln(`sigma') - (ln(`t') - `xb')^2 / (2 * `sigma'^2)
replace `lnf' = ln(1 - normal((ln(`t') - `xb') / `sigma')) if cens == 1

}
end
ml model lf my_lognormal (durat = $X) ()
ml check
ml max, difficult iterate(50)
est store lognormal2
estat ic
matrix results = r(S)
matrix AIC_BIC[4, 1] = results[1, 5] // AIC
matrix AIC_BIC[4, 2] = results[1, 6] // BI

* Exibindo a tabela com os resultados
display "Final AIC and BIC comparison table:"
matrix list AIC_BIC

estout weib weib2 lognormal lognormal2, cells(b(star fmt(3)) se(par fmt(3)))

* Ajustar modelo Weibull
streg $X, dist(weibull) nohr difficult

* Gerar dados para sobrevivência com black = 1
stcurve, survival at(black=1 black=0)

margins, at(priors=(0 1 2 3 4 5 6 7 8 9 10) black=(0 1)) plot

Simulando “duração até uma infecção” a partir da Weibull
cls
clear all
set seed 54321
sca n = 1000
sca T = 5
sca alfa = 1.5
sca beta = -.3
sca rho = .3
set obs `=n*T'
g i = int((_n-1)/T)+1
sort i
by i: g t = _n
g x = rnormal(0,1)
bys i: g c = rnormal(0,2) if t == 1
bys i: replace c = c[_n-1] if mi(c)
g lambda = .
by i: replace lambda = exp(-alfa*(beta*x+c)) if t == 1
g durat = .
by i: replace durat = (-ln(runiform())/lambda)^(1/alfa) if t == 1
forv j = 2/`=T' {
replace lambda = exp(-alfa*(beta*x+c+rho*durat[_n-1])) if t == `j'
replace durat = (-ln(runiform())/lambda)^(1/alfa) if t == `j'
}

by i: g durat_lag = durat[_n-1]
drop if t == 1
replace t = t-1
g cens = runiform() > 0.8
stset durat, fail(cens)
streg x c durat_lag, dist(weibull) time
est sto weib1
streg x durat_lag, dist(weibull) time
est sto weib2
sort i t
bys i: g durat0 = durat[1]
bys i: egen x_mean = mean(x)
xtset i
xtstreg x durat_lag durat0 x_mean if t>1, dist(weibull) time
est sto weib3
estout weib1 weib2 weib3, cells(b(star fmt(3)) se(par fmt(3)))

margins, at(x=(-2 -1 0 1 2) durat_lag=(0 1)) plot

y ∈ N

▶ y é uma contagem Poisson, então y∗ = E(y | controles)



y∗
11 = exp(γy10 + β1 + β2x11 + β3d11 + c1)

y∗
12 = exp(γy11 + β1 + β2x12 + β3d12 + c1)

...
y∗

1T = exp(γy1(T −1) + β1 + β2x1T + β3d1T + c1)
...

y∗
n1 = exp(γyn0 + β1 + β2xn1 + β3dn1 + cn)

y∗
n2 = exp(γyn1 + β1 + β2xn2 + β3dn2 + cn)

...
y∗

nT = exp(γyn(T −1) + β1 + β2xnT + β3dnT + cn)

yi ∼ Poisson(λi), λi = exp(β′xxxi + ci)

L(β) =
n∏

i=1

λyi
i exp(−λi)

yi!
=

n∏
i=1

(exp(β′xxxi + ci))yi exp (− exp(β′xxxi + ci))
yi!

lnL(β) =
n∑

i=1

(
yi(β′xxxi + ci)− exp(β′xxxi + ci)− ln(yi!)

)

L(β, σc) =
n∏

i=1

∫ ∞

−∞

T∏
t=1

λyit
it exp(−λit)

yit!
f(ci) dci

Há uma versão de FE ao estilo C-Logit

▶ Modelos de contagem são usados para variáveis dependentes representando
números inteiros (ex.: número de eventos).

▶ Principais modelos:
▶ Regressão de Poisson
▶ Regressão Binomial Negativa (BN)
▶ Regressão Zero-Inflated Poisson (ZIP)

Regressão de Poisson

Características principais:
▶ Modela contagens de eventos em um intervalo fixo.
▶ Assume que a variável dependente segue uma distribuição de Poisson.
▶ Fórmula do modelo:

log(λi) = Xiβ

Onde:
▶ λi: Média esperada do número de eventos.
▶ Xi: Vetor de covariáveis.
▶ β: Vetor de coeficientes.

Limitações:
▶ Sensível à superdispersão (variância maior que a média).
▶ Não lida bem com excesso de zeros.

160 / 221



Regressão Binomial Negativa (BN)
Características principais:
▶ Modela contagens de eventos com superdispersão.
▶ Relaxa a suposição de que média e variância são iguais.
▶ Fórmula da variância:

Var(Yi) = µi + µ2
i

θ

Onde:
▶ µi: Média esperada.
▶ θ: Parâmetro de dispersão.

Vantagens:
▶ Flexível para lidar com superdispersão.

Limitações:
▶ Não resolve problemas de excesso de zeros.

Regressão Zero-Inflated Poisson (ZIP)
Características principais:
▶ Lida com excesso de zeros.
▶ Assume dois processos geradores de zeros:

▶ Um processo estrutural (sempre zero).
▶ Um processo Poisson.

▶ Fórmula do modelo:

P (Yi = 0) = πi + (1− πi)e−λi

Onde:
▶ πi: Probabilidade do zero estrutural.
▶ e−λi : Probabilidade do zero Poisson.

Vantagens:
▶ Modela zeros estruturais e incidentais separadamente.

Limitações:
▶ Interpretação mais complexa.

Comparação entre Modelos

Modelo Situação Ideal Limitação Principal

Poisson Variância próxima à média Sensível à superdispersão
BN Superdispersão sem excesso de zeros Não lida com excesso de zeros
ZIP Excesso de zeros estruturais Interpretação mais complexa

Escolha do Modelo

Critérios para Escolher o Modelo:
▶ **Superdispersão**:

▶ Compare variância e média da variável dependente.
▶ **Excesso de Zeros**:

▶ Compare a proporção de zeros observados com o esperado.
▶ **Testes Estatísticos**:

▶ Vuong Test: ZIP vs Poisson.
▶ AIC/BIC: Para comparar modelos alternativos.

▶ Cada modelo tem um uso específico:
▶ Poisson: Dados simples sem superdispersão.
▶ BN: Dados com superdispersão.
▶ ZIP: Excesso de zeros estruturais.

▶ Use testes diagnósticos e critérios de seleção para escolher o modelo mais
adequado.

Suplemento da PNAD-C
cls
clear all
tic
infix ///
uf 6-7 ///
capital 8-9 ///
rural 33-33 ///
sexo 95-95 ///
idade 104-106 ///
cor 107-107 ///
naointernet 505-505 ///
avaliapol 559-559 ///
chancevio 597-597 ///
foifurtdado 630-630 ///
vezesfurto 631-632 ///
foiroubado 785-785 ///
vezesroubo 786-787 ///
familia 576-576 ///
amigo 577-577 ///
colega 578-578 ///

vizinho 579-579 ///
guarda 580-580 ///
civil 581-581 ///
pm 582-582 ///
bombeiros 583-583 ///
justica 584-584 ///
forcas_arm 585-585 ///
using "G:/Meu Drive/pnad_c/PNADC_2021_trimestre4.txt", clear
compress
sa "G:/Meu Drive/pnad_c/PNADC_2021_trimestre4.dta", replace
toc

cls
clear all
use "G:/Meu Drive/pnad_c/PNADC_2021_trimestre4.dta"
keep if capital != .
tab vezesfurto vezesroubo, m
g y =.
replace y = 0 if vezesfurto == . & vezesroubo == .
replace y = 1 if (vezesfurto == 1 & vezesroubo == .) | (vezesfurto == . & vezesroubo == 1)
replace y = 2 if (vezesfurto == 2 & vezesroubo == .) | (vezesfurto == . & vezesroubo == 2) \\\

| (vezesfurto == 1 & vezesroubo == 1)
replace y = vezesfurto+vezesroubo if y == .
tab y
replace rural = rural-1
g homem = 2-sexo
g branco = cond(cor==1,1,0)
tab capital, gen(c_)
gl X rural homem branco c_2-c_27
sum y
poisson y $X
nbreg y $X, dispersion(mean)
zip y $X, inflate(_cons)

* Estime o modelo ZIP
zip y $X, inflate(_cons)

* Calcule o preditor linear para ambas as partes do modelo

gen xb_count_rural1 = _b[_cons] + _b[homem]*homem + _b[branco]*branco + _b[rural]*1 ///
+ _b[c_2]*c_2 + _b[c_3]*c_3 + _b[c_4]*c_4 + _b[c_5]*c_5 + _b[c_6]*c_6 + ///
_b[c_7]*c_7 + _b[c_8]*c_8 + _b[c_9]*c_9 + _b[c_10]*c_10 + _b[c_11]*c_11 + ///
_b[c_12]*c_12 + _b[c_13]*c_13 + _b[c_14]*c_14 + _b[c_15]*c_15 + _b[c_16]*c_16 + ///
_b[c_17]*c_17 + _b[c_18]*c_18 + _b[c_19]*c_19 + _b[c_20]*c_20 + _b[c_21]*c_21 + ///
_b[c_22]*c_22 + _b[c_23]*c_23 + _b[c_24]*c_24 + _b[c_25]*c_25 + _b[c_26]*c_26 + ///
_b[c_27]*c_27

gen xb_count_rural0 = _b[_cons] + _b[homem]*homem + _b[branco]*branco + _b[rural]*0 ///
+ _b[c_2]*c_2 + _b[c_3]*c_3 + _b[c_4]*c_4 + _b[c_5]*c_5 + _b[c_6]*c_6 + ///
_b[c_7]*c_7 + _b[c_8]*c_8 + _b[c_9]*c_9 + _b[c_10]*c_10 + _b[c_11]*c_11 + ///
_b[c_12]*c_12 + _b[c_13]*c_13 + _b[c_14]*c_14 + _b[c_15]*c_15 + _b[c_16]*c_16 + ///
_b[c_17]*c_17 + _b[c_18]*c_18 + _b[c_19]*c_19 + _b[c_20]*c_20 + _b[c_21]*c_21 + ///
_b[c_22]*c_22 + _b[c_23]*c_23 + _b[c_24]*c_24 + _b[c_25]*c_25 + _b[c_26]*c_26 + ///
_b[c_27]*c_27

gen xb_inflate_rural1 = _b[inflate:_cons]
gen xb_inflate_rural0 = _b[inflate:_cons]

* Calcule as probabilidades de zeros estruturais
gen p_zero_rural1 = invlogit(xb_inflate_rural1)
gen p_zero_rural0 = invlogit(xb_inflate_rural0)

* Calcule o valor esperado (mu) para a parte de contagem
gen mu_rural1 = exp(xb_count_rural1)
gen mu_rural0 = exp(xb_count_rural0)

* Calcule as probabilidades para Y = 0, 1, 2, 3
gen pr_y0_rural1 = p_zero_rural1 + (1 - p_zero_rural1) * exp(-mu_rural1)
gen pr_y1_rural1 = (1 - p_zero_rural1) * mu_rural1 * exp(-mu_rural1)
gen pr_y2_rural1 = (1 - p_zero_rural1) * (mu_rural1^2 / 2) * exp(-mu_rural1)
gen pr_y3_rural1 = (1 - p_zero_rural1) * (mu_rural1^3 / 6) * exp(-mu_rural1)

gen pr_y0_rural0 = p_zero_rural0 + (1 - p_zero_rural0) * exp(-mu_rural0)
gen pr_y1_rural0 = (1 - p_zero_rural0) * mu_rural0 * exp(-mu_rural0)
gen pr_y2_rural0 = (1 - p_zero_rural0) * (mu_rural0^2 / 2) * exp(-mu_rural0)
gen pr_y3_rural0 = (1 - p_zero_rural0) * (mu_rural0^3 / 6) * exp(-mu_rural0)

* Calcule médias das probabilidades
egen avg_pr_y0_rural1 = mean(pr_y0_rural1)
egen avg_pr_y1_rural1 = mean(pr_y1_rural1)
egen avg_pr_y2_rural1 = mean(pr_y2_rural1)
egen avg_pr_y3_rural1 = mean(pr_y3_rural1)

egen avg_pr_y0_rural0 = mean(pr_y0_rural0)
egen avg_pr_y1_rural0 = mean(pr_y1_rural0)
egen avg_pr_y2_rural0 = mean(pr_y2_rural0)
egen avg_pr_y3_rural0 = mean(pr_y3_rural0)

* Organize os dados para o gráfico
gen y_value = 0
replace y_value = 1 in 2
replace y_value = 2 in 3
replace y_value = 3 in 4

gen pr_rural1 = .
replace pr_rural1 = avg_pr_y0_rural1 in 1
replace pr_rural1 = avg_pr_y1_rural1 in 2
replace pr_rural1 = avg_pr_y2_rural1 in 3
replace pr_rural1 = avg_pr_y3_rural1 in 4

gen pr_rural0 = .
replace pr_rural0 = avg_pr_y0_rural0 in 1
replace pr_rural0 = avg_pr_y1_rural0 in 2
replace pr_rural0 = avg_pr_y2_rural0 in 3
replace pr_rural0 = avg_pr_y3_rural0 in 4

* Gráfico comparativo
twoway (line pr_rural1 y_value, sort) (line pr_rural0 y_value, sort), ///

legend(label(1 "Rural = 1") label(2 "Rural = 0")) ///
title("Comparação de Probabilidades Estimadas para Y (Modelo ZIP)") ///
xlabel(0(1)3) ylabel(, angle(horizontal))

Base de dados de contagem simulada

clear all
set seed 54321
loc n = 100
loc T = 10
loc reps = 50
mat m_poisson = J(`reps', 2, .)
mat m_wooldr = J(`reps', 2, .)
tic
forv rep = 1/`reps' {
clear
set obs `n'
g i = _n
expand `T'
bys i: gen t = _n
sca gamma = .25
sca beta = 1
sort i t
bys i: gen c = runiform()-.5 if _n == 1
bys i: replace c = c[_n-1] if mi(c)
g x = runiform()-.5
g y = rpoisson(runiform()) if t == 1
g lambda = .
forv j = 2/`T' {
bys i: replace lambda = exp(gamma*y[_n-1]+beta*x+c) if t == `j'
bys i: replace y = min(rpoisson(lambda),20) if t == `j'
}

xtset i t
poisson y L.y x
mat m_poisson[`rep',1] = _b[L.y]
mat m_poisson[`rep',2] = _b[x]
// Wooldridge
bys i: gen y0 = y[1]
bys i: egen x_mean = mean(x)
xtpoisson y L.y x y0 x_mean if t > 1, re iter(30)
mat m_wooldr[`rep',1] = _b[L.y]
mat m_wooldr[`rep',2] = _b[x]
}
toc
svmat m_poisson, names(po)
svmat m_wooldr, names(w)
sum po* w*

Modelo de Poisson com Efeitos Fixos

O modelo básico de Poisson é:

Yit ∼ Poisson(µit),

com:
µit = exp(Xitβ + αi + ln(popit)),

onde:
▶ Yit: variável de contagem.
▶ µit: valor esperado de Yit.
▶ Xitβ: combinação linear das covariáveis.
▶ αi: efeito fixo específico da unidade i.
▶ ln(popit): offset (log da exposição popit).

Função de Verossimilhança Completa

A função de verossimilhança para todos os dados é:

L(β, α) =
N∏

i=1

T∏
t=1

µYit
it e−µit

Yit!
.

Substituindo µit:

L(β, α) =
N∏

i=1

T∏
t=1

[exp(Xitβ + αi + ln(popit))]Yit e− exp(Xitβ+αi+ln(popit))

Yit!
.

Problema: Não podemos estimar αi diretamente devido ao problema dos
parâmetros incidentais.

Condicionalização pela Soma Total dos Eventos

A soma total dos eventos para cada unidade é:

Si =
T∑

t=1
Yit.

A probabilidade condicional de Yit, dado Si, elimina αi:

P (Yi1, Yi2, . . . , YiT | Si) =
∏T

t=1 µYit
it∑

{Yit:
∑

t
Yit=Si}

∏T
t=1 µYit

it

.

Aqui, αi desaparece porque afeta µit de forma multiplicativa.

Função de Verossimilhança Condicional

A função de verossimilhança condicional é:

LC(β) =
N∏

i=1

∏T
t=1 [exp(Xitβ + ln(popit))]Yit∑

{Yit:
∑

t
Yit=Si}

∏T
t=1 [exp(Xitβ + ln(popit))]Yit

.

Simplificando:

LC(β) =
N∏

i=1

∏T
t=1 [popit · exp(Xitβ)]Yit∑

{Yit:
∑

t
Yit=Si}

∏T
t=1 [popit · exp(Xitβ)]Yit

.

Interpretação

▶ A condicionalização elimina αi, permitindo estimar β sem viés.
▶ O denominador, que depende de todas as combinações possíveis de Yit que

somam Si, é calculado numericamente.
▶ Os coeficientes β podem ser interpretados como log-taxas relativas (log-rate

ratios).

clear all
import excel "https://petterini.ufsc.br/files/2024/12/contagens.xlsx", sheet("fecam") firstrow
drop if i == "SC"
en i, g(id)
xtset id t
xtpoisson homM t, fe exposure(pop)
est sto pofe
sort id t
bys id: g y0 = homM[1]
bys id: egen t_mean = mean(t)
xtpoisson homM L.homM t y0 t_mean if t > 1, fe iter(30)
est sto wofe
xtpoisson homM L.homM t y0 t_mean if t > 1, re iter(30)
est sto wore
estout pofe wofe wore, cells(b(star fmt(3)) se(par fmt(3)))

y ∈ {a, b, c}

179 / 221



Logit Multinomial

1. Partindo da Função de Utilidade:
▶ O modelo assume que cada indivíduo i escolhe entre J alternativas.
▶ A utilidade da alternativa j para o indivíduo i é dada por:

Uij = β′
jxxxij + ξij ,

onde:
▶ xxxij : vetor de características observáveis da alternativa j e do indivíduo i,
▶ βj : vetor de coeficientes estimáveis específicos da alternativa j,
▶ ξij : componente de erro (não observado).

2. Assumindo Distribuição Gumbel para ξij:
▶ Para simplificar a análise, assume-se que os erros ξij são i.i.d. segundo a distribuição

Gumbel (valor extremo):
F (ξ) = exp(− exp(−ξ)).

▶ Propriedades:
▶ Esta distribuição tem diferenças entre erros bem comportadas, permitindo

derivar probabilidades.
▶ Produz uma função de densidade cumulativa (CDF) para escolhas discretas.

3. Derivando a Probabilidade Logística:
▶ A probabilidade de o indivíduo i escolher a alternativa j é:

Pr(yi = j) = Pr(Uij > Uik, ∀k ̸= j).

▶ Substituímos a utilidade:

Pr(yi = j) = Pr(β′
jxxxij + ξij > β′

kxxxik + ξik, ∀k ̸= j).

▶ Como ξij ∼ Gumbel, as diferenças entre ξij têm distribuição logística, e podemos derivar:

Pr(yi = j) =
exp(β′

jxxxij)∑J

k=1 exp(β′
kxxxij)

.

4. Normalizando a Utilidade:
▶ A utilidade é identificável apenas até uma constante comum.
▶ Para resolver essa indeterminação, é comum normalizar a utilidade da categoria de referência

(j = 1):
Ui1 = 0 ⇒ β1 = 000.

▶ As probabilidades ajustadas são:

Pr(yi = 1) = 1
1 +

∑J

k=2 exp(β′
kxxxij)

,

Pr(yi = j) =
exp(β′

jxxxij)
1 +

∑J

k=2 exp(β′
kxxxij)

, j = 2, . . . , J.

Modelo Logit Multinomial com Efeitos Fixos

O modelo logit multinomial é usado para modelar uma variável de escolha discreta
Yit com J categorias (j = 1, 2, . . . , J).

P (Yit = j) = exp(Vijt)∑J
k=1 exp(Vikt)

,

com:
Vijt = Xijtβ + αij ,

onde:
▶ Xijt: covariáveis específicas da alternativa j no tempo t.
▶ β: coeficientes estimáveis.
▶ αij : efeito fixo específico da unidade i para a alternativa j.

Problema dos Efeitos Fixos

A função de verossimilhança completa seria:

L(β, α) =
N∏

i=1

T∏
t=1

J∏
j=1

P (Yit = j)dijt ,

com:
P (Yit = j) = exp(Xijtβ + αij)∑J

k=1 exp(Xiktβ + αik)
.

Problema:
▶ Muitos αij a estimar (N × J).
▶ Viés dos parâmetros incidentais.

Condicionalização para Eliminar αij

Condicionamos na soma total de escolhas por alternativa j para cada indivíduo i:

Sij =
T∑

t=1
dijt, onde dijt = 1 se Yit = j, e 0 caso contrário.

A probabilidade condicional das escolhas {Yit}, dado Sij , elimina αij :

P ({Yit} | Si1, . . . , SiJ) =
∏T

t=1
∏J

j=1 exp(Xijtβ)dijt∑
{Yit:Sij}

∏T
t=1

∏J
j=1 exp(Xijtβ)dijt

.

Função de Verossimilhança Condicional

A função de verossimilhança condicional é:

LC(β) =
N∏

i=1

∏T
t=1

∏J
j=1 exp(Xijtβ)dijt∑

{Yit:Sij}
∏T

t=1
∏J

j=1 exp(Xijtβ)dijt
.

Aqui:
▶ exp(Xijtβ): representa a utilidade estimada da alternativa j para a unidade i

no tempo t.
▶ O denominador normaliza a probabilidade condicionada, eliminando αij .

Interpretação e Implementação

▶ Os efeitos fixos αij desaparecem devido à condicionalização.
▶ A verossimilhança condicional permite estimar β consistentemente.
▶ Aplicações incluem escolha de produtos, decisões políticas, transporte, etc.
▶ Estimável via pacotes estatísticos que suportam modelos logit multinomial de

efeitos fixos.

cls
clear all
bcuse keane
xtset id year
mlogit status educ expwc expersq black, baseoutcome(1)
*est sto m1
femlogit status educ expwc expersq black, group(id) baseoutcome(1)
*est sto m2
*estout m1 m2, cells(b(star fmt(3)) se(par fmt(3)))
g d2 = cond(status==2,1,0)
g d3 = cond(status==3,1,0)
bys id: g d20 = d2[1]
bys id: g d30 = d3[1]
bys id: egen educ_mean = mean(educ)
bys id: egen expwc_mean = mean(expwc)
bys id: egen expersq_mean = mean(expersq)
femlogit status L.status educ expwc expersq black d20 d30 educ_mean expwc_mean expersq_mean if year > 81, group(id) baseoutcome(1)

Em termos de dinâmica, a defasagem deve ser escrita com um jogo de dummies, mas precisa ter um comando de

efeitos aleatórios

cls
clear all
use "G:/Meu Drive/pnad_c/PNADC_2021_trimestre4.dta"
keep if capital != . | naointernet == .
tab naointernet
replace rural = rural-1
g homem = 2-sexo
g branco = cond(cor==1,1,0)
tab capital, gen(c_)
global X i.rural homem branco idade c_2-c_27
mlogit naointernet $X, baseoutcome(7)
margins rural, at(homem=1 branco=1 idade=(20 30 40 50 60 70 80) ///

c_2=1 c_3=0 c_4=0 c_5=0 c_6=0 c_7=0 c_8=0 c_9=0 c_10=0 ///
c_11=0 c_12=0 c_13=0 c_14=0 c_15=0 c_16=0 c_17=0 c_18=0 ///
c_19=0 c_20=0 c_21=0 c_22=0 c_23=0 c_24=0 c_25=0 c_26=0 c_27=0) ///
predict(outcome(5))

marginsplot, ///
title("Probabilidade de naointernet = 5 por Idade e Rural") ///
xlabel(20(10)80) ylabel(, angle(horizontal)) ///
ytitle("Probabilidade Pr(naointernet=5)") xtitle("Idade") ///
legend(label(1 "Rural = 0") label(2 "Rural = 1"))

y1 < y2 < y3

▶ Os modelos **Logit Ordenado** e **Probit Ordenado** são usados para
variáveis dependentes ordinais.

▶ Baseiam-se na ideia de uma **utilidade latente** contínua (U∗
i ), que

determina as categorias observáveis (yi).
▶ A estrutura subjacente permite inferir o impacto de variáveis independentes

nas probabilidades das categorias ordinais.

Estrutura de Utilidade Latente
Modelo Subjacente:

U∗
i = Xiβ + ϵi,

onde:
▶ U∗

i : utilidade latente (não observada),
▶ Xi: vetor de covariáveis observáveis,
▶ β: vetor de coeficientes estimáveis,
▶ ϵi: termo de erro.

Categorias Observáveis:

yi = j se µj−1 < U∗
i ≤ µj , j = 1, 2, . . . , J.

▶ µj : pontos de corte estimados, separando as categorias de yi.

Diferença entre Logit e Probit Ordenado
Distribuição do Termo de Erro ϵi:
▶ Logit Ordenado: ϵi ∼ Logística(0, π2/3)

F (ϵi) = exp(ϵi)
1 + exp(ϵi)

.

▶ Probit Ordenado: ϵi ∼ N (0, 1)

F (ϵi) = Φ(ϵi),

onde Φ(·) é a CDF da normal padrão.

Probabilidades:

Pr(yi = j) = F (µj −Xiβ)− F (µj−1 −Xiβ).

- Para o Logit Ordenado: F (·) é a CDF logística. - Para o Probit Ordenado: F (·)
é a CDF normal.

Pontos de Corte e Probabilidades

Pontos de Corte (µj):
▶ Os pontos de corte são estimados junto com os coeficientes β.
▶ Representam os limites entre as categorias observadas (yi) na escala de U∗

i .

Probabilidades:

Pr(yi = j) = F (µj −Xiβ)− F (µj−1 −Xiβ).

- Pr(yi = j): probabilidade de yi estar na categoria j. - Depende da posição
relativa de Xiβ em relação aos pontos de corte µj e µj−1.

c1 c2 c3

P (y = 1) P (y = 2) P (y = 3) P (y = 4)
y∗

f(y∗)

Interpretação de β

▶ Os coeficientes β capturam o efeito das covariáveis Xi na utilidade latente U∗
i .

▶ O impacto de β em Pr(yi = j) é indireto, pois depende:
▶ Dos valores de Xiβ,
▶ Da posição relativa dos pontos de corte µj .

▶ Em vez de mudanças absolutas, β representa mudanças relativas na propensão
de yi mudar de categoria.

cls
clear all
use "G:/Meu Drive/pnad_c/PNADC_2021_trimestre4.dta"
keep if capital != . | chancevio == .
tab chancevio
replace rural = rural-1
g homem = 2-sexo
g branco = cond(cor==1,1,0)
tab capital, gen(c_)
tab avaliapol chancevio, m
drop if avaliapol == . | avaliapol == 9 | chancevio == . | chancevio == 9
replace avaliapol = 6-avaliapol
replace chancevio = 5-chancevio
tab avaliapol chancevio, m
global X i.avaliapol rural homem branco idade c_2-c_27
ologit chancevio $X
margins avaliapol, at(rural=0 homem=1 branco=1 idade=30 ///

c_2=0 c_3=0 c_4=0 c_5=0 c_6=0 c_7=0 c_8=0 c_9=0 c_10=0 ///
c_11=0 c_12=0 c_13=0 c_14=0 c_15=0 c_16=0 c_17=0 c_18=0 ///
c_19=0 c_20=0 c_21=0 c_22=0 c_23=0 c_24=0 c_25=0 c_26=0 c_27=0) ///
predict(outcome(4)) predict(outcome(1))

marginsplot, ///
title("Probabilidades de chancevio = 4 e chancevio = 1 por avaliapol") ///
xlabel(1(1)5, format(%2.0f)) ylabel(, angle(horizontal)) ///
ytitle("Probabilidade") xtitle("avaliapol") ///
legend(label(1 "Pr(chancevio = 4)") label(2 "Pr(chancevio = 1)"))

cls
clear all
use "G:/Meu Drive/pnad_c/PNADC_2021_trimestre4.dta"
drop if capital == . | familia == . | familia == 9
tab familia, m
replace familia = 6-familia
replace rural = rural-1
g homem = 2-sexo
g branco = cond(cor==1,1,0)
tab capital, gen(c_)
global X rural homem branco i.idade c_2-c_27
ologit familia $X
margins idade, at(rural=0 homem=1 branco=1 ///

c_2=0 c_3=0 c_4=0 c_5=0 c_6=0 c_7=0 c_8=0 c_9=0 c_10=0 ///
c_11=0 c_12=0 c_13=0 c_14=0 c_15=0 c_16=0 c_17=0 c_18=0 ///
c_19=0 c_20=0 c_21=0 c_22=0 c_23=0 c_24=0 c_25=0 c_26=0 c_27=0) ///
predict(outcome(5))

marginsplot, ///
title("Probabilidade de familia = 5 por idade") ///
xlabel(, format(%2.0f)) ylabel(, angle(horizontal)) ///
ytitle("Probabilidade Pr(familia = 5)") xtitle("Idade")

cls
clear all
use "G:/Meu Drive/pnad_c/PNADC_2021_trimestre4.dta"
drop if capital == . | bombeiros == . | bombeiros == 9
tab bombeiros, m
replace bombeiros = 6-bombeiros
replace rural = rural-1
g homem = 2-sexo
g branco = cond(cor==1,1,0)
tab capital, gen(c_)
global X rural homem branco i.idade c_2-c_27
ologit bombeiros $X
margins idade, at(rural=0 homem=1 branco=1 ///

c_2=0 c_3=0 c_4=0 c_5=0 c_6=0 c_7=0 c_8=0 c_9=0 c_10=0 ///
c_11=0 c_12=0 c_13=0 c_14=0 c_15=0 c_16=0 c_17=0 c_18=0 ///
c_19=0 c_20=0 c_21=0 c_22=0 c_23=0 c_24=0 c_25=0 c_26=0 c_27=0) ///
predict(outcome(5))

marginsplot, ///
title("Probabilidade de bombeiros = 5 por idade") ///
xlabel(, format(%2.0f)) ylabel(, angle(horizontal)) ///
ytitle("Probabilidade Pr(bombeiros = 5)") xtitle("Idade")

Simulação de pesquisa de satisfação

cls
clear all
set seed 54321
set obs 1000
g renda = rnormal(3000,800) // Renda média de 3000 com desvio padrão de 800
g idade = round(runiform()*47+18) // Idade entre 18 e 65 anos
g frequencia_uso = ceil(runiform()*3) // Frequência de uso: 1 (baixa), 2 (média), 3 (alta)
* Gerar a variável latente z
g z = .5*(renda/1000)-.1*idade+.8*frequencia_uso+rnormal(0,.5)
* Definir os pontos de corte para gerar satisfação
g satisfacao = .
replace satisfacao = 1 if z <= -1.5
replace satisfacao = 2 if z > -1.5 & z <= -0.5
replace satisfacao = 3 if z > -0.5 & z <= 0.5
replace satisfacao = 4 if z > 0.5 & z <= 1.5
replace satisfacao = 5 if z > 1.5
* Limpar a variável latente
drop z

ologit satisfacao renda idade i.frequencia_uso

* Interpretação dos resultados
margins, at(renda=(2000(1000)5000) idade=(20(10)60) frequencia_uso=(1 2 3))

Em termos de dinâmica, a defasagem deve ser escrita pelo próprio y

y em fronteiras

Fronteira Estocástica

▶ A fronteira estocástica é usada para medir a eficiência técnica (ET ) de
unidades produtivas.

▶ O modelo considera uma fronteira de produção máxima e distingue:
▶ Ineficiência Técnica (Ui): desvios sistemáticos da fronteira.
▶ Erro Aleatório (Vi): choques imprevisíveis.

▶ Para dados em painel, a análise incorpora a dimensão temporal:

ln Yit = f(Xit; β)− Uit + Vit.

Componentes do Modelo

▶ Yit: output observado para a unidade i no tempo t.
▶ f(Xit; β): fronteira de produção máxima, especificada como:

▶ Cobb-Douglas: f(Xit; β) = β0 +
∑k

j=1 βj ln Xijt,
▶ Translog: Inclui termos quadráticos e de interação.

▶ Uit: termo de ineficiência técnica (Uit ≥ 0).
▶ Vit ∼ N(0, σ2

v): termo de erro aleatório.

Ineficiência Constante ou Variável no Tempo

Ineficiência Constante no Tempo:

Uit = Ui, ETit = exp(−Ui).

Ineficiência Variável no Tempo:
▶ Modelo de decaimento exponencial (Battese e Coelli, 1992):

Uit = Ui · exp(−ηt), ETit = exp(−Ui · exp(−ηt)).

▶ Modelo geral:
Uit ∼ Truncated-Normal(µit, σ2

u).

Função de Verossimilhança

Ineficiência Constante no Tempo:

f(ϵi1, . . . , ϵiT ) =
∫ ∞

0

T∏
t=1

ϕ(ϵit + Ui; σ2
v) · exp

(
−Ui

σu

)
dUi,

onde:
▶ ϵit = Vit − Ui: resíduo.
▶ ϕ(·): densidade da normal padrão.

Ineficiência Variável no Tempo:

L =
N∏

i=1

T∏
t=1

∫ ∞

0
f(ln Yit | Xit, Uit, β) · g(Uit)dUit.

Estimativa de Eficiência Técnica

▶ A eficiência técnica (ETit) é dada por:

ETit = exp(−Uit).

▶ O termo Uit é estimado condicionalmente aos resíduos ϵit:

E[Uit | ϵit] = σ2
u

σ2

(
ϵit +

ϕ
( ϵit

σ

)
Φ
(
− ϵit

σ

)) ,

com:
σ2 = σ2

u + σ2
v .

Vantagens e Desvantagens

Vantagens:
▶ Distingue ineficiência técnica (Uit) de erros aleatórios (Vit).
▶ Permite estimar mudanças na eficiência ao longo do tempo.
▶ Adequado para dados em painel com múltiplas observações por unidade.

Desvantagens:
▶ Requer supostos fortes sobre distribuições (Uit e Vit).
▶ Sensível à especificação da função de produção.

208 / 221



Eficiência em produção
clear all
set seed 54321
* Configuração: 20 firmas e 10 períodos de tempo
set obs 20
g firm_id = _n
expand 10
bys firm_id (firm_id): g year = _n
* Gerar inputs
g labor = rnormal(50, 10)
g capital = rnormal(100, 20)
* Gerar termo de ineficiência técnica (U_i >= 0)
g inefficiency = abs(rnormal(1, 0.5))
bys firm_id (year): replace inefficiency = inefficiency[_n-1] if _n > 1
* Gerar termo de erro aleatório (V_it)
g error = rnormal(0, 0.2) // Erro aleatório
* Gerar output (Y_it) com uma fronteira Cobb-Douglas
g ln_output = 2+.7*ln(labor)+.3*ln(capital)-inefficiency+error
g output = exp(ln_output)
* Criar dummies de tempo para capturar efeitos fixos no tempo
tab year, gen(time_)
keep firm_id year labor capital output
* Aplicar o comando xtfrontier no Stata
g lnY = ln(output)
g lnL = ln(labor)
g lnK = ln(capital)
xtset firm_id year
xtfrontier lnY lnL lnK, ti
xtfrontier lnY lnL lnK, tvd
predict efficiency, te
kdensity efficiency
graph box efficiency, over(year, sort(ascending))

Eficiência em Custos

▶ A eficiência em custos mede a capacidade de uma unidade produtiva de
minimizar seus custos para um dado nível de output.

▶ Compara os custos observados (Cit) com os custos mínimos (C∗
it).

▶ Essencial para:
▶ Diagnosticar ineficiências.
▶ Identificar economias de custo.
▶ Melhorar a competitividade.

Função de Custo
A função de custo representa o custo mínimo necessário para produzir um dado
nível de output:

C∗
it = f(Yit, Pit; β),

onde:
▶ C∗

it: custo mínimo,
▶ Yit: output produzido,
▶ Pit: vetor de preços dos insumos,
▶ β: parâmetros da função de custo.

O custo observado (Cit) pode ser maior que C∗
it devido a:

▶ Ineficiência Técnica (Uit): desvio sistemático,
▶ Erro Aleatório (Vit): choques não sistemáticos.

Modelo de Fronteira de Custo
A fronteira estocástica para custos é modelada como:

ln Cit = f(ln Yit, ln Pit; β) + Uit + Vit,

onde:
▶ f(ln Yit, ln Pit; β): função de custo mínima (log-linearizada),
▶ Uit ≥ 0: termo de ineficiência de custo,
▶ Vit ∼ N(0, σ2

v): erro aleatório.
A eficiência de custo é calculada como:

ECit = exp(−Uit),

onde:
▶ ECit = 1: eficiência perfeita (custo mínimo),
▶ ECit < 1: presença de ineficiência de custo.

Distribuição de Ineficiência (Uit)

▶ O termo Uit é não-negativo (Uit ≥ 0) e segue uma distribuição:
▶ Truncada normal: Uit ∼ Truncated-Normal(µ, σ2

u),
▶ Exponencial: Uit ∼ Exponential(λ).

▶ Vit é simétrico, enquanto Uit induz assimetria em ln Cit.
▶ A separação de Uit e Vit é feita pela maximização da função de

verossimilhança.

Estimativa de Eficiência em Custos

A eficiência técnica condicional (Uit) é estimada usando a expectativa condicional:

E[Uit | ϵit] = σ2
u

σ2

(
ϵit +

ϕ
( ϵit

σ

)
Φ
(
− ϵit

σ

)) ,

onde:
▶ ϵit = ln Cit − f(ln Yit, ln Pit; β),
▶ σ2 = σ2

u + σ2
v .

A eficiência de custo é dada por:

ECit = exp(−Uit).

Aplicações e Interpretação

Aplicações Práticas:
▶ Análise de eficiência em setores industriais.
▶ Benchmarking entre empresas.
▶ Identificação de economias de escala e escopo.

Interpretação:
▶ ECit = 1: Unidade opera no custo mínimo.
▶ ECit < 1: Unidade apresenta custos excessivos em relação ao mínimo.
▶ Comparação entre ECit ao longo do tempo identifica mudanças na eficiência.

Eficiência em custos

clear all
set seed 54321
* Configuração: 20 firmas e 10 períodos de tempo
set obs 20
gen firm_id = _n
expand 10
bys firm_id (firm_id): gen year = _n
* Gerar inputs
gen labor = rnormal(50, 10) // Trabalho
gen capital = rnormal(100, 20) // Capital
gen output = rnormal(200, 50) // Output produzido
* Gerar termo de ineficiência técnica (U_i >= 0)
gen inefficiency = abs(rnormal(0.1, 0.05)) // Ineficiência nos custos
bys firm_id (year): replace inefficiency = inefficiency[_n-1] if _n > 1
* Gerar termo de erro aleatório (V_it)
gen error = rnormal(0, 0.02) // Erro aleatório nos custos
* Preço dos inputs
gen price_labor = rnormal(20, 5) // Preço do trabalho
gen price_capital = rnormal(10, 2) // Preço do capital
* Função de custo Cobb-Douglas (C_it = f(Y, P_L, P_K) + U_i + V_it)
gen ln_cost = 2 + 0.6 * ln(output) + 0.3 * ln(price_labor) + 0.1 * ln(price_capital) + inefficiency + error
gen cost = exp(ln_cost)

* Criar dummies de tempo
tab year, gen(time_)
* Variáveis para o modelo
gen lnY = ln(output)
gen lnPL = ln(price_labor)
gen lnPK = ln(price_capital)
gen lnC = ln(cost)
* Manter variáveis relevantes
keep firm_id year lnC lnY lnPL lnPK
* Configurar painel
xtset firm_id year
* Estimar o modelo com ineficiência constante no tempo
xtfrontier lnC lnY lnPL lnPK, ti
* Estimar o modelo com ineficiência variável no tempo
xtfrontier lnC lnY lnPL lnPK, tvd
* Predizer eficiência de custo
predict efficiency_cost, te
graph box efficiency_cost, over(year, sort(ascending)) ///

title("Eficiência de Custo por Ano") ytitle("Eficiência de Custo")

217 / 221



Acrescentando dinâmica
clear all
set seed 54321
set obs 20
gen firm_id = _n
expand 10
bys firm_id (firm_id): gen year = _n
gen labor = rnormal(50, 10)
gen capital = rnormal(100, 20)
gen output = rnormal(200, 50)
gen price_labor = rnormal(20, 5)
gen price_capital = rnormal(10, 2)
gen cost = exp(2+.6*ln(output)+.3*ln(price_labor)+.1*ln(price_capital)+abs(rnormal(0, 0.2)))
bysort firm_id (year): replace cost = exp(0.2 * ln(cost[_n-1]) + ///

(2+.6*ln(output)+.3*ln(price_labor)+.1*ln(price_capital)+abs(rnormal(0,.2)))) ///
if _n > 1

* Gerar log-transformações
gen lnC = ln(cost)
gen lnY = ln(output)
gen lnPL = ln(price_labor)
gen lnPK = ln(price_capital)
bysort firm_id (year): gen lnC_lag = lnC[_n-1] if _n > 1
* Configurar painel
xtset firm_id year
* Estimar modelo de fronteira de custo com lag
xtfrontier lnC lnY lnPL lnPK lnC_lag, ti
* Estimar modelo com ineficiência variável no tempo
xtfrontier lnC lnY lnPL lnPK lnC_lag, tvd
* Predizer eficiência de custo
predict efficiency_cost, te

